2 Kurvenscharen

Wir werden zunächst grundlegende Tabellentechniken erlernen, insbesondere die verschiedenen Arten der Zelladressierung, dann Abbildungen erstellen und Schieberegler einsetzen (in Abschn. 2.2) und schließlich Kurvenscharen berechnen und in Abbildungen darstellen, z. B. kreisförmige Wellenfronten von bewegten akustischen Sendern. Wir berechnen außerdem Erwartungswerte für Exponentialfunktionen und simulieren musikalische Effekte mit Kosinusfunktionen. Didaktisches Ziel: Variablen in Funktionen und Formeln sollen mit ihren Namen eingesetzt werden.

2.1 Einleitung: Sin, Cos, Exp

Tabellentechnik
In diesem Kapitel sollen Sie üben, wie Zellen adressiert, Abbildungen erstellt und formatiert sowie Schieberegler eingesetzt werden. Das wird Ihnen leicht fallen, wenn Sie sich schon mit EXCEL auskennen und wissen, wie man Formeln in Zellen schreibt. Wenn Sie weniger geübt sind, dann gehen Sie zunächst die Rezepte in der Grundübung Schritt für Schritt durch, sehen bei Bedarf in der EXCEL-Hilfe nach und besorgen sich schließlich ein für Sie passendes Lehrbuch über EXCEL-Techniken.

Benötigte und geübte Excel-Techniken sind:

- relative und absolute Adressierung,
- direkte und indirekte Adressierung,
- Benennung von Zellen und Zellbereichen,
- Diagramme erstellen und
- Schieberegler einsetzen.

Wir wenden dabei die Tabellenfunktionen SIN, COS, EXP, ABS an.

D. Mergel, *Physik mit Excel und Visual Basic*,
DOI 10.1007/978-3-642-37857-7_2

Wir üben zunächst verschiedene Arten der Zelladressierung. Unser Ziel wird aber sein, Formeln möglichst mathematisch zu schreiben, also mit Buchstaben für Variablen. Dazu müssen einzelne Zellen, Zellbereiche in Zeilen oder Spalten und zweidimensionale Zellbereiche (Matrizen) mit Namen versehen werden. Das wird in den einzelnen Übungen schrittweise eingeführt und in Übung 2.8 noch einmal zusammengefasst.

Physikalische Aufgaben
Wir simulieren die Schallausbreitung einer bewegten Quelle mit einer Schar von Kreisen, deren Radien mit der Schallgeschwindigkeit wachsen und deren Mittelpunkt mit der Geschwindigkeit der Quelle wandert. Wir werden die Schallmauer eindrucksvoll mit einem Schieberegler durchbrechen.

Akustische Signale werden mit einer Summe von Kosinusfunktionen nachgebildet, sodass man Oberwellen und Schwebungen sichtbar machen kann.

Die Exponentialfunktion scheint eine langweilige Form zu haben, birgt aber Überraschungen, die sich in der Strom-Spannungs-Kennlinie einer Diode zeigen. Bei der Berechnung der mittleren Lebensdauer verschiedener radioaktiver Substanzen lassen wir den Leser in eine numerische Falle tappen.

2.2 Grundübung in Tabellenkalkulation

Wir tabellieren zwei Funktionen und stellen sie in einer Abbildung dar. Wir üben die Grundtechniken der Tabellenrechnung: Absolute und relative Zell-Adressierung, Zellen mit Namen versehen, grafische Darstellung, Verknüpfung von Text und Variablen, Schieberegler. Der Leser soll durch die Übungen in die Lage versetzt werden, sich ein ausführliches Lehrbuch über EXCEL-Techniken nach seinen Bedürfnissen auszuwählen.

Aufgabe
Führen Sie diese Grundübung Schritt für Schritt durch!

2.2.1 Kosinus und Parabel

Wir tabellieren zwei Funktionen und stellen sie in einem Diagramm dar. Die Funktionen sind eine Parabel mit zwei Parametern *a* und *b:*

$$f_1(x) = a \cdot x^2 + b \tag{2.1}$$

und eine Kosinusfunktion mit einem Parameter *C:*

$$f_2(x) = C \cdot \cos(x) \tag{2.2}$$



Abb. 2.1 **a** (links, T) Tabellierung der Parabel, Gl. 2.1, und des Kosinus, Gl. 2.2, für $x = 3$ bis 8. Die Zeilen 4 (leer) und 12 bis 34 sind ausgeblendet. Die Zellinhalte in schräger Ausrichtung geben Formeln in darunterstehenden Zellen wieder, deren Inhalt fett gedruckt ist. **b** (rechts) Standardmäßig von Excel vorgeschlagene Darstellung der beiden Kurven, hier grau statt farbig

Vorgeschlagener Tabellenaufbau

Der Tabellenaufbau wird in Abb. 2.1a (T) wiedergegeben. Die unabhängigen Variable x (vor Γ) und die Funktionswerte $f(x)$ und cos(x) (unterhalb Γ) stehen in parallelen Spalten, A, B und C. Die Parameter a, b sowie der Abstand dx zwischen den Stützstellen werden in den Zellen B1:B3 festgelegt. Diese Zellen sollen mit ihren Namen (in A1:A3) aufgerufen werden.

In der Tabelle von Abb. 2.1a (T) stehen in Zeile 7 und 5 in schräger Ausrichtung die Formeln der fett markierten Zellen in Zeile 11 bzw. 6.

Die grafische Darstellung der beiden Kurven, so wie sie von Excel standardmäßig vorgeschlagen wird, sehen Sie in Abb. 2.1b. Die vorgeschlagenen Farben werden hier aber nur grau wiedergegeben. Die Legenden „f(x)" und „cos(x)" sollen aus den Zellen B9 bzw. C9 von Abb. 2.1a übernommen werden. Formatierungen des Diagramms werden im nächsten Abschn. 2.2.2 besprochen.

Zuweisungen von Formeln zu Zellen werden wir künftig im Text folgendermaßen beschreiben: A11 = [=A10 + dx]. Wir müssen nämlich unterscheiden, ob das Gleichheitszeichen in die Zelle geschrieben wird oder nicht. Bei den Ausdrücken A9 = [x] und A10 = [3] wird kein Gleichheitszeichen in die Zelle geschrieben. [x] wird dann als Text aufgefasst und [3] als Zahl.

Fragen

Von A10 bis A11 in Abb. 2.1a erhöht sich x um dx = 0,23. Warum geht der nächste Sprung von 3,23 auf 8,75?[1]

Wo wird dx festgelegt?[2]

---

[1] Die Zeilen 12 bis 34 sind ausgeblendet. Der Sprung geht über 23 Fortschritte von dx.

[2] Der Wert für dx wird in B3 festgelegt. B3 wird mit dem Namen in A3 belegt.

Stützstellen

Gl. 2.1 und 2.2 geben kontinuierliche Funktionen wieder. In Tabellen werden die Funktionswerte y aber nur an einer endlichen Anzahl von x-Werten, x_i, berechnet. Die Punkte (x_i, y_i) nennt man Stützstellen der Funktion. In Abb. 2.1a (T) und b werden die Funktionen mit 26 Stützstellen dargestellt. In den meisten Übungen haben benachbarte Stützstellen denselben Abstand, in Abb. 2.1a als $\mathrm{d}x = 0{,}23$ in B3 definiert. Man spricht dann von „äquidistanten Stützstellen".

Zellen mit Namen versehen

Im Bereich A1:B3 von Abb. 2.1a werden drei Parameter definiert, die überall in der Tabelle abgerufen werden können. Wir schreiben in A1:A3 die Namen, die diese Parameter haben sollen: a und b für die Koeffizienten der Funktion f_1 und dx für die Abstände der x-Werte, für die die Funktionen berechnet und später im Diagramm dargestellt werden sollen. In B1:B3 tragen wir dann die gewünschten Zahlenwerte ein. Zur Verknüpfung der drei Namen in Spalte A mit den drei Werten in Spalte B markieren wir den Bereich A1:B3 und klicken uns durch (Excel 2010):

Formeln/Definierte Namen/Aus Auswahl erstellen.

Ein Aufruf erscheint: Namen erstellen aus Spalte an linker Seite? Ja, das hat der Agent richtig erkannt, und wir bestätigen mit Ok. Mehr zum Namens-Manager siehe Abschn. 2.8.

Wenn wir jetzt in eine Zelle irgendwo in der Tabelle [=a] schreiben und die Enter-Taste drücken, dann steht sofort der entsprechende Zahlenwert in der Zelle, in unserem Beispiel 0,17. Wenn wir den Wert in Zelle B1 ändern, dann ändert sich auch sofort der Wert in der Zelle mit [=a].

Argumente in gleichen Abständen

Wir können uns jetzt daranmachen, die x-Werte zu bestimmen, an denen die beiden Funktionen berechnet werden sollen. Das machen wir im Spaltenbereich A10:A35. Vorher schreiben wir ordentlich als Überschrift in A9 = [x]. In die erste Zelle des Bereichs schreiben wir den gewünschten Anfangswert von x, also z. B. A10 = [3]. In die zweite Zelle, A11, schreiben wir A11 = [=A10 + dx]. Wir können „A10" Buchstaben für Buchstaben reinschreiben oder nach „=" einfach die Zelle A10 anklicken und „+dx" dazuschreiben. Nach Enter erscheint sofort 3,23 in der Zelle, weil dx = B3 = 0,23.

Wir klicken die gerade beschriebene Zelle A11 an, greifen das kleine Quadrat in der rechten unteren Ecke der Zellumrandung, den „Henkel", mit dem Mauszeiger und ziehen bei gedrückter Maustaste hinunter bis Zelle A35. Nach Enter erscheinen alle x-Werte bis 8,75 in dem gewünschten Abstand dx = 0,23. Wir haben auf diese Weise äquidistante (in gleichen Abständen) Argumente erzeugt.

Aufgabe

Verändern Sie den Inhalt der Zelle B3, mit „dx" benannt! Sofort sollten sich alle x-Werte in den Zellen A11:A35 anpassen.

Tabellierung der Parabel

Jetzt machen wir uns daran, die Parabel zu tabellieren. Dazu müssen wir nur in die Zelle B10 eine Formel eintippen, die f_1 (Gl. 2.1) entspricht:

$$\mathrm{B10} = \left[\mathrm{a} * \mathrm{A10}\hat{}2 + \mathrm{b}\right].$$

Die Koeffizienten a und b können wir mit ihren Namen einschreiben, da wir diese Namen ja vorher den entsprechenden Zellen zugeordnet haben, a = B1 = [0,23], b = B2 = [−3,3]. Für x setzen wir den Wert aus A10 ein, entweder durch buchstabengetreues Einschreiben oder einfach durch Klicken auf die Zelle A10.

Die Variable aus A10 muss quadriert werden. Das geschieht mit dem Operator ^ (Potenzoperator). Man muss auf die Taste mit dem ^-Zeichen drücken und dann auf die gewünschte Potenz, hier also „2"; erst danach erscheint „^2". Der Rest ist einfach: Wir tippen „+b". Nach ENTER sollte der Funktionswert in der gerade beschriebenen Zelle erscheinen. Wenn nicht, dann überprüfen Sie, ob Sie die obige Formel tatsächlich mit allen Buchstaben, Zahlen, * und + eingegeben haben. Die fertige Formel in B11 wird in der Zelle B7 als Text in schräger Ausrichtung wiedergegeben.

Wir klicken die Zelle B11 erneut an und doppelklicken auf die rechte untere Ecke, den „Henkel". Dann wird der Zelleninhalt sofort bis zur 35. Zeile fortgeschrieben, also für alle Zellen, die in der benachbarten Spalte, hier in Spalte A, bereits ohne Unterbrechung beschrieben sind. In der letzten Zelle B35 steht jetzt B35 = [=a*A35^2 + b], also nicht [=… A10…], wie in der ersten von uns beschriebenen Zelle. Die Formel wird eben folgerichtig fortgeschrieben und nicht einfach buchstabengetreu kopiert. Der Zellbezug ist *relativ*. Damit haben wir unsere erste Funktion programmiert. Wenn wir die Werte von a und b in B1 und B2 verändern, dann verändern sich auch sofort die Funktionswerte in Spalte B.

Tabellierung des Kosinus mit absoluter und relativer Zelladressierung

Die zweite Funktion berechnen wir in Spalte C. Ordentlicherweise schreiben wir in Zelle C9 den Text „cos(x)" ohne Gleichheitszeichen, also C9 = [cos(x)], damit wir später noch wissen, was wir gemacht haben. Wir benötigen die Konstante C und die Kosinusfunktion. Den gewünschten Wert für C schreiben wir in C8. Dann schreiben wir wieder nur eine Formel in die erste Zelle des Funktionsbereichs, also in C10, und kopieren diese Formel wie vorher beschrieben die Spalte C hinunter. In C10 schreiben wir zunächst nur:

$$\mathrm{C10} = [= \cos(\mathrm{A10})]$$

Die Schreibweise bedeutet, dass in die Zelle C10 der Text „=Cos(A10)" geschrieben wird. Das Gleichheitszeichen muss mit in die Zelle geschrieben werden.

COS ist eine in EXCEL definierte Tabellenfunktion und wird deshalb von uns in KAPITÄLCHEN gesetzt. Die Argumente der Tabellenfunktionen, in unserem Fall A10, werden in runde Klammern gesetzt. Beim Hinunterziehen dieser Zelle wird sofort die Funktion COS(x) bis hinunter nach C35 berechnet, wobei die x-Werte fortlaufend aus der jeweils selben Zeile in Spalte A genommen werden, so wie wir das für die erste Funktion in Spalte B schon kennengelernt haben.

▶ Begriffe, die in Excel definiert sind, schreiben wir im Text mit Kapitälchen.

Fragen

Was ist der Unterschied zwischen C10 = [Cos(A10)] und C10 = [=Cos(A10)]?[3]

Die Amplitude C schreiben wir in die Zelle C8. Wenn wir C10 ergänzen zu „=C8 * cos(A10)“ und hinunterziehen, dann gibt es Probleme. In der letzten Zelle C35 steht dann „=C33 * cos(A35)“, was wir natürlich in diesem Fall nicht beabsichtigt haben. Der Vorfaktor soll immer der Wert in C8 sein. Dazu müssen wir die Formel in C10 abändern, indem wir vor die 8 ein Dollarzeichen setzen:

C10 = [=C$8*COS(A10)]

Jetzt weiß Excel, dass es in diesem Tabellenblatt auf Zeile 8 zugreifen soll, und in Zelle C35 steht C35 = [=C$8*Cos(A35)]. Das $-Zeichen gibt an, dass der Zellbezug *absolut* ist.

▶ Ψ *Der Dollar macht's absolut.*

Stichworte für die Excel-Hilfe: Adressierung, relativ und absolut.

2.2.2 Grafische Darstellung der Funktionen

Nachdem wir die Tabellen so schön erstellt haben, wollen wir die Funktionen auch grafisch darstellen. Dazu setzen wir den Zeiger auf eine leere Zelle weit weg von den beschriebenen Zellen und klicken auf: Einfügen/Diagramme/ (Abb. 2.2a), und innerhalb der Karte Diagramme auf Punkt. Es erscheint ein Unterfenster „XY“, in dem wir Nur mit Datenpunkten wählen und mit Enter bestätigen. Es wird ein leeres Diagramm eingefügt.

Nachdem wir bei aktiviertem Diagramm in der Registerkarte Entwurf/ (siehe Abb. 2.2b) Daten/Daten auswählen/Hinzufügen/ geklickt haben, öffnet sich das Fenster in Abb. 2.3, dessen Zeilen wir ausfüllen. Das geschieht am besten durch Klicken auf das Tabellensymbol.

In der Tabelle markieren wir Bereiche, z. B. A10:A35, für Werte der Reihe X und bestätigen mit OK, ähnlich für Werte der Reihe Y. Wenn wir das für die erste Funktion gemacht haben, dann klicken wir in der Karte Daten auswählen noch einmal auf Hinzufügen und wiederholen die Schritte für die Funktion f_2. Das Diagramm hat sich jetzt mit Punkten gefüllt und sieht etwa wie in Abb. 2.1b aus.

[3] Bei C10 = [Cos(A10)] steht ein Text in der Zelle, bei C10 = [=Cos(A10)] eine Formel.



Abb. 2.2 Registerkarten, die für Diagramme wichtig sind, nach EINFÜGEN/DIAGRAMME/XY im Start-Menü (EXCEL 2010): **a** (links) EINFÜGEN und AUSWAHL FORMATIEREN, ganz links in der Startleiste, um ein Diagramm einzufügen bzw. ein Element des Diagramms zu formatieren; nach Klicken auf ▾ erscheint Abb. 2.4b. **b** (rechts) ENTWURF/DATEN AUSWÄHLEN (Karte Abb. 2.3 erscheint) FORMAT, z. B. ganz rechts in der Startleiste, um die Größe des Diagramms zu bestimmen



Abb. 2.3 Einfügen von Datenreihen in ein Diagramm. Der Reihenname wird am besten aus der Tabelle entnommen, *nicht* als Text eingegeben

Aufgabe
Verändern Sie die Parameter a, b, dx und C und den Anfangswert für x in A10 von Abb. 2.1a (T) und beobachten Sie, wie sich Tabelle und Diagramm ändern!

▶ **Alac** Das ist ja cool! Das Diagramm lebt!

▶ **Tim** Einmal erstellt, immer aktuell!

Formatieren des Diagramms
Wir können das Aussehen unseres Diagramms nach unseren Vorstellungen verändern, sodass es z. B. wie in Abb. 2.4a aussieht.

Nach Anklicken des Diagramms und der Registerkarte (DIAGRAMM/FORMAT) in der Gruppe DIAGRAMMTOOLS wurden folgende Komponenten geändert:

- Größe (7 cm hoch, 8 cm breit) (in der Bearbeitungszeile ganz rechts, siehe Abb. 2.2b), Befehl wurde noch nicht ausgeführt.

Außerdem nach Anklicken des betreffenden Elementes des Diagramms oder nach Auswahl im Fenster der Abb. 2.4b (in der Registerkarte FORMAT):

- Kastenumrahmungen (kein Rahmen für den Diagrammbereich, schwarzer Rahmen für die Zeichnungsfläche),
- Strichstärke und Farbe der horizontalen und vertikalen Achse (schwarz, 1 Pt),
- Zahlenformate für die Achsenbeschriftungen (vertikale Achse formatieren, horizontale Achse formatieren),
- Skalierung der x-Achse (HORIZONTAL (WERT) ACHSE), nämlich von 3 bis 8,
- Form, Größe und Farbe der Datenpunkte (REIHEN „f(x)" und REIHEN „cos(x)"),
- Linienfarbe für die erste Kurve, geändert von KEINE LINIE zu EINFARBIGE LINIE.. SCHWARZ.

Alles das geschieht, indem man das Diagramm anklickt und dann im Fenster in der Registerleiste ganz links (Abb. 2.2a) aussucht, was man formatieren will. Zunächst steht DIAGRAMMBEREICH im Fenster. Nach Öffnen der Liste durch Klicken auf ▾ tauchen aber alle Elemente des Diagramms auf (Abb. 2.4b).

Nach der gewünschten Auswahl betätigt man AUSWAHL FORMATIEREN und macht dann im Fenster, das sich daraufhin öffnet, weiter.

Verkettung von Text und Variablen für die Legende in Abbildungen Ψ *„Text" & Variable*

Im ursprünglichen Diagramm haben wir als Namen für die Kurven die Zelleninhalte in B9 und C9 von Abb. 2.1 (T) angegeben, in denen einfache Texte stehen. Im neuen Diagramm werden die Funktionen aber mit den tatsächlich gewählten Parametern bezeichnet, mit Werten aus den Zellen B6 und C6. Die Formeln für diese Zellen sind vom Typ Ψ *„Text"&Variable* und stehen über der Zelle in schräger Ausrichtung, z. B.

$$\mathrm{B5} = [= \mathrm{C8}\&\,„\cdot \cos(x)\text{“}], \quad \text{ausgegeben als} \quad [5 \cdot \cos(x)]$$

$$\mathrm{C5} = [\mathrm{a}\&\,„\cdot x^2 + \text{“}\&\mathrm{b}], \quad \text{ausgegeben als} \quad [0{,}21 \cdot \mathrm{x}^2 + -3{,}3]$$



Abb. 2.4 **a** (links) Abb. 2.1b nach Formatierung der Achsen und der Kurven. **b** (rechts) Fenster nach Auswahl, durch Anklicken, des Diagramms, der Registerkarte FORMAT und des Zeichens ▾

Mit der ersten Formel wird der Inhalt der Zelle C8 mit dem darauffolgenden angeführten Text verknüpft. Soll Text eingefügt werden, so muss er in Anführungszeichen gesetzt werden, wie oben „· cos(x)“. Das Zeichen für den Punkt „·“ erhält man über die Registerkarte EINFÜGEN/SYMBOLE. Der Verkettungsoperator ist &. Wenn der Inhalt von Zelle C8 verändert wird, dann werden die Ausgabe in C6 und die Legende im Diagramm sofort angepasst.

Oftmals muss man Werte runden. Wenn z. B. in Zelle F7 = 1/3 steht, dann wird dort nur 0,33 angezeigt. Wird F7 aber in die Legende eingesetzt, dann erscheint 0,33333333. Setzt man RUNDEN(F7, 2) ein, dann wird 0,33 eingesetzt: [=„x=“&RUNDEN(F7; 2)] ergibt als Zellinhalt [x = 0,33].

▶ Ψ *„Text“ & Variable*

2.2.3 Schieberegler

Schieberegler
Die folgenden Angaben gelten für EXCEL 2010. In EXCEL 2016 gibt es ein deutschsprachiges Menü mit folgenden Entsprechungen: LINKEDCELL = ZELLVERKNÜPFUNG, SMALLCHANGE = SCHRITTWEITE, LARGECHANGE = SEITENWECHSEL..

Mit einem Schieberegler (SCROLL BAR) können Sie den Inhalt einer Zelle (LINKEDCELL) mit ganzen Zahlen zwischen 0 und 32.767 (= $2^{15}-1$) beschreiben Die Einstellungen legen Sie im Blatt EIGENSCHAFTEN (Abb. 2.5c) fest: Für LINKEDCELL geben Sie die Zelladresse an, die beschrieben werden soll. Der Wertebereich kann mit MIN und MAX eingeschränkt werden. SMALLCHANGE und LARGECHANGE bestimmt die Schrittweiten, wenn Sie auf die Pfeile an den Rändern bzw. neben den Reiter des Schiebereglers klicken.

Wenn Sie andere Zahlen brauchen, dann müssen Sie diese aus den Zahlen in der LINKEDCELL mit einer Formel in einer anderen Zelle ableiten.

In späteren Kapiteln wird Folgendes wichtig sein: Man kann mit dem Schieberegler eine VBA-Routine verbinden (z. B. in Abschn. 5.8). Eine Routine SUB SCROLLBAR1_CHANGE() … END SUB wird jedes Mal ausgeführt, wenn der Schieberegler SCROLLBAR1 betätigt wird. Dazu klicken Sie auf CODE ANZEIGEN (Abb. 2.5b) und wählen im linken Drop-down-Fenster des vba-Editors den Namen des gewünschten Schiebereglers und im rechten Drop-down-Fenster CHANGE (oder eine andere Aktion, bei der die Routine ausgelöst werden soll).

Zellinhalte werden mit Schiebereglern verändert (ohne Zahlen einzutippen)
Wir können noch eindrucksvoller mit Kurven spielen, wenn wir die Parameter durch Schieberegler verändern und dabei gleichzeitig die Diagramme beobachten. Um Schieberegler einzuführen, betätigen wir ENTWICKLERTOOLS/EINFÜGEN/ACTIVEX-STEUERELEMENTE und erhalten die Register in Abb. 2.5a (EXCEL 2010).

Wenn die Registerkarte in Ihrem Menüband nicht auftaucht, dann müssen Sie das Menüband anpassen, indem Sie bei Datei\Optionen\Menüband ein Häkchen bei ☑ Entwicklertools machen.

Wir benötigen den Schieberegler als ActiveX-Steuerelement und klicken auf das Symbol für den Schieberegler in der oberen Zeile ganz rechts (Abb. 2.5a) und ziehen dann mit der Maus an der gewünschten Stelle in der Tabelle ein Rechteck auf. In Abb. 2.5b, die eine Fortsetzung von Abb. 2.1a ist, wurde das in den Zellen I1, I2 und I3 gemacht. Jetzt schaltet sich der Entwurfsmodus ein, und wir können den Schieberegler konfigurieren. In Abb. 2.5c wird die Eigenschaftenliste für Excel 2010 gezeigt.

In Abb. 2.6 (T) stehen die ersten drei Zeilen der Tabelle, von der Ausschnitte schon in Abb. 2.1 (T) und Abb. 2.5b gezeigt wurden.



Abb. 2.5 **a** (links) Registerkarte nach Betätigen von Entwicklertools/Einfügen/; im Untermenü ActiveX-Steuerelemente wird ein Schieberegler in der oberen Reihe ganz rechts gelistet, siehe Pfeil in den Kreis. **b** (Mitte) In I1, I2 und I3 wurden drei Schieberegler eingebaut, durch Klicken auf das Steuerelement und Aufziehen in einem Tabellenbereich. **c** (rechts) Menü, mit dem die Eigenschaften des Schiebereglers (ScrollBar) festgelegt werden. Es erscheint nach Klicken auf Eigenschaften in b. Wichtige Parameter: LinkedCell, Minimum und Maximum der Werte. In Excel 2016 erscheint ein deutschsprachiges Menü

	A	B	E	H	I
1	**a**	0,17	*=H1/100*	17	
2	**b**	-0,3	*=(H2-50)/10*	47	
3	**dx**	1	*=H3/100*	100	

Abb. 2.6 (T) Die Parameter *a*, *b* und d*x* werden in Spalte B aus den mit den Schiebereglern verbundenen Zellen in Spalte H berechnet. Die Formeln in Spalte B stehen in Spalte E in Kursivschrift. Diese Abbildung zeigt Zellen, die auch in Abb. 2.1a und 2.5b zu sehen sind

Fragen

Welche Zahlen (Bereich und Abstand der Zahlen zueinander) können nach den Angaben in Abb. 2.5c in B1 von Abb. 2.6 (T) erscheinen?[4]

Wir aktivieren den Schieberegler in I1 und klicken auf EIGENSCHAFTEN. In EXCEL 2010 erscheint das Menü in Abb. 2.5c. Wenn es nicht erscheinen sollte, dann haben Sie irgendwie den ENTWURFSMODUS ausgeschaltet. Schalten Sie ihn wieder ein! Wir legen fest, dass die Zelle H1 (LINKEDCELL) beschrieben werden soll und dass die Zahlen zwischen 0 (MIN) und 100 (MAX) liegen sollen. Wir schalten dann den ENTWURFSMODUS durch Anklicken des entsprechenden Schalters in Abb. 2.5a aus und bewegen den Schieber in I1 mit der Maus. Sofort erscheint eine Zahl in H1. Entsprechend den Anweisungen für I1 verfahren wir mit den Schiebereglern in I2 und I3.

Wenn der ENTWURFSMODUS eingeschaltet wird, dann können bereits bestehende Steuerelemente verändert oder neue hinzugefügt werden. Wenn der ENTWURFSMODUS ausgeschaltet ist, dann können die Steuerelemente betätigt werden.

Aus den Zahlen in H1:H3 werden jetzt die Parameter a, b und dx in B1:B3 berechnet, siehe Abb. 2.6. Wenn jetzt einer der Schieberegler betätigt wird, dann passen sich die Werte in den Zellen und die Kurven in Abb. 2.4a sofort an. Ein Schieberegler kann ganze Zahlen von 0 bis 32.767 ($=2^{15}$) erzeugen. Wird ein anderer Bereich gewünscht, dann muss in einer anderen Zelle der Wert der LINKEDCELL entsprechend geändert werden. In Abb. 2.6 (T) wird mit B2 = [=(H2 − 50)/10] der Wert der LINKEDCELL H2, dessen Wert in Abb. 2.5a (T) zwischen 0 und 100 eingestellt wird, in B2 in den Bereich −5 bis 5 verschoben, mit Abständen von 0,1.

Wenn man den Reiter mit dem Cursor fasst und schiebt, dann wird die Ausgabe in der verbundenen Zelle verändert. SMALLCHANGE gibt die Sprünge (nach links oder nach rechts) an, die die Zahlen machen, wenn man auf den (linken bzw. rechten) Rand des Schiebereglers klickt. LARGECHANGE gibt die Sprünge an, wenn man in den Schieberegler links oder rechts des Reiters klickt. Probieren Sie es aus!

Umgekehrt gilt: Wenn man den Inhalt einer LINKEDCELL (ZELLVERKNÜPFUNG in EXCEL 2016) ändert, dann verschiebt sich der Reiter des zugehörigen Schiebereglers.

2.2.4 Zusammenfassung: Zellbezüge

▶ Ψ *Der Dollar macht's absolut.*

[4] MIN = 0; MAX = 100; In B1 wird durch 100 geteilt, B1 = [=H1/100]. Es können also Zahlen von 0 bis 1 im Abstand von 1/100 = 0,01 erscheinen.

Zellbezüge, absolute, relative, indirekte
Die Formeln in einer Zelle, z. B. C3 = [=A4], C3 = [=$A4], C3 = [=A$4], C3 = [=$A$4], ergeben zunächst alle dasselbe Ergebnis: In die aktuelle Zelle C3 wird der Wert der Zelle A4 eingeschrieben. Werden diese Formeln jedoch in eine andere Zelle kopiert, dann verändern sie sich. In D4 steht dann z. B.: [=B5], [=$A5], [=B$4], [=$A$4]. Ein $ vor einer Spalten- oder Reihenbezeichnung besagt, dass diese Bezeichnung beim Kopieren festgehalten wird; es handelt sich um einen *absoluten* Zellbezug. Fehlt das $, dann wird die Bezeichnung um den Spalten- oder Zeilenabstand zwischen alter und neuer Zelle weitergezählt; es handelt sich um einen *relativen* Zellbezug.

Die Tabellenfunktion INDIREKT(ZELLE) erwartet als Argument eine Zelladresse. Sie schreibt den Inhalt der Zelle mit dieser Adresse in die aktuelle Zelle. Ein Beispiel: Mit A4 = [=INDIREKT(A5); A5 = [X7]; X7 = [3,4] steht der Wert 3,4 in A4.

Formel in nur eine Zelle schreiben und dann kopieren, Γ-Aufbau einer Tabelle
In diesem Unterabschnitt wird eine Übersicht über verschiedene Arten der Zeladressierung gegeben, die in den vorhergehenden Abschnitten geübt wurden. Zur Illustration nehmen wir eine Schar von vier Geraden:

$$f_i(\mathrm{x}) = a_i \cdot x + b_i \text{ mit } i = 1,\ 2,\ 3,\ 4, \tag{2.3}$$

deren Koordinaten erzeugt werden sollen, indem nur eine einzige Zelle mit einer Formel beschrieben werden soll, die dann durch „Ziehen" des „Henkels" in einen großen Tabellenbereich kopiert wird.

Die vier Funktionen des Typs der Gl. 2.3 können z. B. wie in Abb. 2.7 (T) in einer Tabelle berechnet werden, die klar im Γ-Aufbau („Gamma"-Aufbau) in drei Bereiche gegliedert ist. (A1:E3) (Zeilenbereiche oberhalb Γ für den Abstand d*x*

	A	B	C	D	E	F	G	H	I
1	**dx**	0,1					C7		
2	**a**	2,26	2,42	0,64	4,81		1,0241	=INDIREKT(G1)	
3	**b**	0,56	0,78	4,05	0,21				
4	=A6+$B$1						D	15	
5	**x**	**G.1**	**G.2**	**G.3**	**G.4**		4,63276	=INDIREKT(G4&H4)	
6	0,0	0,56	=B$2*$A6+B$3						
7	**0,1**	0,79	1,02	4,12	0,69				
15	0,9	2,59	2,96	4,63	4,54				
16	1,0	2,82	3,21	4,70	5,02	=E$2*$A16+E$3			

Abb. 2.7 (T) Vier Geraden G_1 bis G_4 im Bereich B6:E16; Typischer Γ-Aufbau

der Stützstellen und die Parameter der Funktionen), (A6:A16) (Spaltenbereich links von Γ für die unabhängige Variable) und (B6:E16) (Matrixbereich unterhalb Γ für die Berechnung der Funktionen aus den Parametern und der unabhängigen Variablen).

Fragen

Wie viele Werte enthält der Spaltenbereich A6:A16 in Abb. 2.7 (T) und warum sieht man davon nur vier Werte?[5]

Die unabhängige Variable x steht in 11 Punkten im Spaltenbereich A6:A16. Die Parameter a und b stehen in den Zeilenbereichen B2:E2 bzw. B3:E3. Die je 11 Funktionswerte für G_1, G_2, G_3 und G_4 sollen in den Spaltenbereichen B6:B16, C6:C16 usw. abgespeichert werden. Für die 4×11 Berechnungen soll aber nur in eine Zelle eine Formel geschrieben werden, hier in B6, die dann ohne weitere Änderung in den großen Tabellenbereich B6:E16 kopiert werden kann und z. B. in E16 wie in F16 gezeigt erscheint.

Man erhält die Pfeile in Abb. 2.7 (T), die angeben, aus welchen Zellen Informationen in die aktuelle Zelle eingehen, mit der Folge FORMELN/FORMELÜBERWACHUNG/SPUR ZUM VORGÄNGER.

Absolute und relative Zelladressierung

Ein Beispiel für einen *gemischten* Zellenbezug (absolut und relativ in einer Adresse) steht in Zelle B6 von Abb. 2.7 (T) (Formeltext in C6):

$$\text{B6} = \left[\text{=B\$2*\$A6+B\$3}\right]$$

Die Formel wurde mit absoluten und relativen Zellbezügen geschrieben. Wird sie in Zelle E16 kopiert, dann wird sie von EXCEL in

$$\text{E16} = \left[\text{=E\$2*\$A16+E\$3}\right]$$

verändert.

Die Bezüge, vor denen ein $ steht, werden nicht verändert, die anderen Bezüge passen sich an die neue Zellposition an.

Indirekter Zellbezug

Es gibt eine dritte Art des Zellbezugs, mit der Tabellenfunktion INDIREKT. Im Argument dieser Funktion wird auf eine Zelladresse verwiesen, deren Inhalt dann als Ergebnis ausgegeben wird. Ein Beispiel steht in Zelle G2 von Abb. 2.7 (T): G2 = [=INDIREKT(G1)]. Da in der Zelle G1 der Text „C7“ steht, G1 = [C7], wird der Inhalt der Zelle C7 angezeigt.

[5] A6:A16 enthält 11 Zellen. Die Zeilen 8 bis 14 sind ausgeblendet.

Das Argument für Indirekt in Zelle G5 wird aus dem Inhalt von zwei Zellen zusammengesetzt G5 = [=Indirekt(G4&H4)]. Auf diese Weise kann man einzelne Werte aus großen Tabellen stärker ins Blickfeld rücken, weil die Zelladressen im Argument von Indirekt einfach geändert werden können, hier also die Zellen G4 und H4. Beispiele dazu finden Sie in Abschn. 8.8.2, laufende Lupe bei der Auswertung von Röntgenbeugungsdaten. Das Argument von Indirekt kann auch mit Ψ *„Text“&Variable* erstellt werden.

Zellbereiche mit Namen ansprechen

Eine elegante Möglichkeit, Zellbereiche aufzurufen, bieten benannte Zellbereiche, mit denen die Tabellenformel der mathematischen Form in Gl. 2.3 angepasst werden kann. Dazu erhält der Spaltenbereich A6:A16 im Namens-Manager den Namen „*x*“ und die Zeilenbereiche B2:E2 und B3:E3 die Namen „a“ bzw. „b“, siehe die Abb. 2.8 (T). Näheres zum Namensmanager siehe Abschn. 2.8.

Die Formel in Zelle B6 lautet dann: I6 = $[=a{*}x+b]$. Wird sie in eine andere Zelle kopiert, dann bleibt der Text gleich, bezieht sich aber auf den Eintrag im Spaltenbereichen x in derselben Zeile und die Einträge in den Zeilenbereichen a und b in derselben Spalte, entsprechend den Pfeilen in Abb. 2.8 (T) auf die Zelle D16.

In Abschn. 2.8 wird der Aufruf von benannten Bereichen noch einmal vertieft im Zusammenhang dargestellt.

Benennung von Zellbereichen

$$\left[=a^{*}x+b\right] \text{ statt } \left[=B\$\,2^{*}\$\,A6+B\$\,3\right]$$

Einzelne Zellen, Spaltenbereiche, Zeilenbereiche und Matrizen können mit Namen versehen werden und dann in Formeln und als Argument in Tabellenfunktionen mit diesen Namen aufgerufen werden. Das geschieht mit dem Namens-Manager, den Sie mit Formeln/Namens-Manager aufrufen können. Näheres zum Namens-Manager siehe Abschn. 2.8.

	A	B	C	D	E	F	G	H	I
1	dx	0,1					C7		
2	a	2,26	2,42	0,64	4,81		1,0241	=INDIREKT(G1)	
3	b	0,56	0,78	4,05	0,21				
4	=A6+dx						D	15	
5	x	G.1	G.2	G.3	G.4		4,63276	=INDIREKT(G4&H4)	
6	0,0	0,56	=a*x+b						
7	0,1	0,79	1,02	4,12	0,69				
15	0,9	2,59	2,96	4,63	4,54				
16	1,0	2,82	3,21	4,70	5,02	=a*x+b			

Abb. 2.8 (T) Ähnlich Abb. 2.7 (T), die Formeln aber mit einem benannten Spaltenbereich namens x für die unabhängige Variable und zwei Zeilenbereichen namens a und b formuliert

Machen Sie ausgiebig Gebrauch von dieser Möglichkeit, sodass Sie Formeln und Funktionen so schreiben können, wie es mathematisch üblich ist, also Z. B. [=*A**SIN(*k**x*)] statt [=A$1* SIN($A5*B$2)]. Das gilt für eine Funktionenschar, deren Amplituden *A* und Wellenzahlvektoren *k* in zwei Zeilenbereichen mit Namen „A" und „k" und die unabhängige Variable *x* in einem Spaltenbereich mit Namen „x" gespeichert werden.

2.2.5 Was haben wir gelernt und wie geht es weiter?

▶ **Alac** EXCEL geht ja echt super einfach. Tabellenrechnung ist ein Kinderspiel. Das war mir noch gar nicht klar.

▶ **Tim** Na ja. Haben wir vielleicht nur Schmalspurkenntnisse für ganz spezielle Aufgaben erworben?

▶ **Mag** Wir haben ein weites Gebiet auf einem schmalen Pfad schnell durchschritten. Er ist der schnelle Weg zum Erfolg, jedenfalls bei den Aufgaben, die wir bearbeiten wollen.

▶ **Tim** Ist es nicht besser, gründlicher zu lernen, damit man nicht verloren ist, wenn die Aufgaben etwas anders gestellt werden?

▶ **Alac** Das kann man doch alles durch Probieren herausbekommen.

▶ **Mag** Ja, Herumprobieren ist eine gute Strategie. Das sollten Sie bei allen Konstruktionen machen, die Sie noch nicht kennen. Gehen Sie aber auch in eine Buchhandlung oder in die Bibliothek und blättern Sie Bücher zum Erlernen von EXCEL durch. Stöbern Sie dabei längs des Pfades, den Sie in dieser Übung kennengelernt haben. Dann merken Sie schnell, welches der Bücher die Abläufe so erläutert, wie Sie sie am besten verstehen. *Dieses Buch sollten Sie sich dann kaufen!*

2.3 Summe von vier Kosinusfunktionen

Wir bilden die Summe von vier Kosinusfunktionen. Wenn die Frequenzen Vielfache einer Grundfrequenz sind, dann entstehen Funktionen, die den Zeitsignalen von Klängen entsprechen. Wenn die Abstände zwischen benachbarten Frequenzen gleich groß sind, dann entstehen Schwebungen. Für die Summenformel des Kosinus gilt die Besenweisheit: Ψ *Cos plus Cos gibt Mittelwert mal halbe Differenz.*

2.3.1 Kosinus (G)

Kosinus mit T und t_0 oder mit ω und ϕ

In Abb. 2.9a wird eine Kosinusfunktion dargestellt. Ihre Amplitude A, Periodendauer T und Zeitverschiebung t_0 werden durch Pfeile markiert. Diese Kenngrößen lassen sich unmittelbar an einer grafischen Darstellung ablesen.

In mathematischen Lehrbüchern werden die Parameter Kreisfrequenz ω und Nullphase φ bevorzugt (siehe Gl. 2.4 in der Box), die sich in die Periodendauer und die Zeitverschiebung umrechnen lassen.

Kosinusfunktion

Die Kosinusfunktion im Zeitbereich wird meist folgendermaßen angegeben:

$$f(t) = A \cdot \cos(\omega t + \phi_0) \tag{2.4}$$

mit der Amplitude A, der Kreisfrequenz ω und der Nullphase ϕ_0. In einem Diagramm lassen sich aber direkt außer A die Periodendauer T und die Zeitverschiebung t_0 ablesen, siehe Abb. 2.9a. Die Funktion mit diesen Parametern lautet dann:

$$f(t) = A \cdot \cos\left(2\pi\left(\frac{t - t_0}{T}\right)\right) \tag{2.5}$$

Die Kenngrößen ω, ϕ_0 und T, t_0 lassen sich ineinander umrechnen. Die Amplitude A ist in beiden Fällen gleich. Der Wert für π kann mit der Tabellenfunktion PI() abgerufen werden.



Abb. 2.9 **a** (links) Eine Kosinusfunktion mit den Parametern Amplitude A, Zeitverschiebung t_0 und der Periodendauer T, die man alle in der grafischen Darstellung ablesen kann. **b** (rechts, T) Aufbau einer Tabelle zur Berechnung von vier Kosinusfunktionen mit den Parametern Amplitude A (in B1:E1), Kreisfrequenz ω (in B2:E2) und Nullphase φ (in B3:E3); im Bereich B6:E36 steht immer dieselbe Formel, die auf drei benannte Zeilenbereiche und einen benannten Spaltenbereich zugreift. Die Pfeile auf C36 erhält man mit FORMELN/FORMELÜBERWACHUNG/SPUR ZUM VORGÄNGER

Fragen

Wie groß sind Amplitude A, Periodendauer T und Zeitverschiebung t_0 in Abb. 2.9a?[6]

Wie berechnet man ω und $_0$ aus T und t_0?[7]

Wie groß sind die Periodendauern der Kosinusfunktionen in Abb. 2.9b?[8]

▶ **Alac** Die Kurve in Abb. 2.9a soll ein Kosinus sein?

▶ **Mag** Ja.

▶ **Tim** Die sieht doch eher wie ein Sinus aus.

▶ **Mag** Das ist richtig. Da wir aber den Kosinus in der Box mit einer Nullphase φ_0 versehen haben, Gl. 2.4, verschwindet der Unterschied zum Sinus. Der Sinus ist dann einfach ein Kosinus mit der Nullphase π/2 bzw. mit der Zeitverschiebung $t_0 = T/4$.

2.3.2 Obertöne

Schwingungen einer Saite



Abb. 2.10 **a** (links) Oszilloskopbild eines Mikrofonsignals einer schwingenden Gitarrensaite als Funktion der Zeit (Dank an Norbert Renner, Universität Duisburg-Essen), Zeiteinheit des Gitters = 5 ms. **b** (rechts) Grundschwingung und zweite Oberschwingung einer an beiden Seiten eingespannten Saite

[6] $A = 2$; $T = 4$; $t_0 = 1$.

[7] $\omega = 2\pi/T$; $\varphi_0 = -\omega \cdot t_0$.

[8] $T = 2\pi/\omega$. $T_{1\varphi} = 2\pi/1 = 6{,}28$; $T_2 = 2\pi/2 = 3{,}14$; $T_3 = T_4 = 2\pi/3 = 2{,}93$ in Zeiteinheiten.

Es geht mit Physik los. Wie schwingt eine Gitarrensaite?

▶ **Mag** Schauen Sie sich das Mikrofonsignal in Abb. 2.10a an. Wie würden Sie das Signal beschreiben?

▶ **Alac** Nun, es schlägt periodisch aus und dazwischen zappelt es wild durcheinander.

▶ **Mag** Ja und nein. Das Signal wiederholt sich mit einer Grundfrequenz; aber Klänge zappeln nicht, sie kommen durch Oberschwingungen zustande. Abb. 2.10a ist die Aufzeichnung des Klangs einer Gitarrensaite. Welche Frequenz hat die Grundschwingung in diesem Fall?

▶ **Tim** Die Periodendauer beträgt 7,5 ms, was einer Frequenz von 133 Hz entspricht. Aber was sind Oberschwingungen?

▶ **Mag** Betrachten wir die Schwingungen einer Gitarrensaite. Wie schwingt eine Saite?

▶ **Tim** Sie wird sinusförmig ausgelenkt wie in Abb. 2.10b.

▶ **Mag** Ja. Abb. 2.10b legt nahe, dass es nur diskrete Frequenzen gibt, mit denen eine Saite schwingen kann. Man kann sie bestimmen, wenn die Randbedingungen berücksichtigt werden.

▶ **Tim** Der Sinus muss durch Null gehen, wo die Saite eingespannt ist.

▶ **Alac** Zwischendurch kann sie auch durch Null gehen.

▶ **Mag** Genau, die Nulldurchgänge nennt man Knoten. Die Randbedingungen erzwingen, dass die Saitenlänge ein Vielfaches n der halben Wellenlänge $\lambda/2$ sein muss. Es gilt also $\boldsymbol{l} = \boldsymbol{n} \cdot (\boldsymbol{\lambda}/2)$ oder $\boldsymbol{f} = \boldsymbol{c}/(2\boldsymbol{l}) \cdot \boldsymbol{n}$, wobei c die Schallgeschwindigkeit auf der Saite ist. Die möglichen Frequenzen sind Vielfache der Grundfrequenz $\boldsymbol{c}/(2\boldsymbol{l})$. Die zugehörigen Schwingungen werden Oberschwingungen genannt. Jetzt lassen Sie mich meine Frage wiederholen: Wie beschreiben Sie das Mikrofonsignal?

▶ **Alac** Ich vermute, es ist die Summe von Oberschwingungen.

▶ **Mag** Richtig, es ist die Summe der erlaubten Schwingungen mit individuellen Amplituden. Im Falle der Saite, die an beiden Seiten festgeklemmt ist, sind die Frequenzen Vielfache der Grundfrequenz. In einem Fall, wo ein Ende des schwingenden Mediums frei ist, wie zum Beispiel bei einem einseitig eingespannten schwingungsfähigen Blatt, sind die Frequenzen ungerade Vielfache der Grundfrequenz.

▶ **Tim** Jetzt sollen wir das Mikrofonsignal nachbilden?

▶ **Mag** Ja, aber in einem allgemeineren Zusammenhang als Summe von vier Kosinusfunktionen, deren Frequenzen bestimmten Bedingungen genügen.

Von Anfang an: Klarer Aufbau der Tabelle!

Aufgabe
Erstellen Sie eine Tabellenrechnung, in der vier Kosinusfunktionen mit frei wählbaren Frequenzen und Amplituden aufsummiert werden. Die Funktionen sollen von $t = 0$ bis $t = 8$ s für 801 Stützpunkte berechnet werden. Den grundsätzlichen Γ-Aufbau einer geeigneten Tabelle sehen Sie in Abb. 2.9b.

▶ **Mag** Das Problem der Obertöne ist ein Spezialfall dieser Aufgabe. Erinnern Sie sich an die Struktur des Rechenmodells, die wir für Kurvenscharen vorgeschlagen haben?

▶ **Tim** Die Funktionen sollen in einem Matrixbereich der Breite 4 (Anzahl der Funktionen) und der Höhe 801 (Anzahl der Datenpunkte) berechnet werden. Man schreibt die gewünschten Werte für Amplituden und Frequenzen in Zeilen über den Matrixbereich und die unabhängige Variable in eine Spalte links vom Matrixbereich.

▶ **Mag** Gut gelernt! Welches ist hier die unabhängige Variable? In welchen Einheiten werden die Frequenzen angegeben?

▶ **Tim** Die unabhängige Variable ist die Zeit. Die Frequenzen werden mit dem Kehrwert der Zeit angegeben.

▶ **Mag** Genau. Zeit und Frequenz sind reziprok zueinander. Diesen Zusammenhang werden Sie später für eine Teilaufgabe brauchen. Als dritter Parameter tritt noch eine Nullphase auf. Der allgemeine Kosinus schreibt sich dann wie Gl. 2.4.

▶ **Mag** Machen Sie sich an die Aufgabe! Vor Ihnen liegen drei Unteraufgaben, die mit dem einen Rechenmodell behandelt werden können, welches wir gerade besprochen haben: Obertöne, Schwebungen und die Summenregel für den Kosinus. Gestalten Sie die Tabelle mit einem Γ-Aufbau wie z. B. in Abb. 2.9b.

Obertöne

Die Frequenzen von Obertönen einer beidseitig eingespannten Saite sind ganzzahlige Vielfache einer Grundfrequenz.

Aufgabe

Variieren Sie die Grundfrequenz, die Amplituden und die Phasen, nehmen Sie mal geradzahlige, mal ungeradzahlige Vielfache der Grundfrequenz und beobachten Sie, wie sich das Summensignal ändert! Zwei Beispiele sehen Sie in Abb. 2.11a und b. Die Funktionen werden an 801 Stützpunkten berechnet.

Eine mögliche Tabellenorganisation im typischen Γ-Aufbau sehen Sie in Abb. 2.12 (T). Sie sollen die Tabelle so organisieren, dass Sie nur eine Zelle mit



Abb. 2.11 **a** (links) Summe aus einem Grundton mit $\omega_1 = 1$ und drei Obertönen mit $2\times$, $3\times$, $4\times\omega_1$. Die Zeiteinheit ist 1 s, wenn die Werte für ω mit 1/s angegeben werden. **b** (rechts) Summe aus einem Grundton mit $\omega_1 = 0{,}5$ und drei Obertönen mit $3\times$, $5\times$, $7\times\omega_1$. Die Zeiteinheit ist 1 s

	A	B	C	D	E	F	G	H	I	J
1		=w.1	=3*w.1	=5*w.1	=7*w.1					
2	**A**	0,23	0,75	0,63	0,72					
3	**w**	**1,00**	**2,00**	**3,00**	**4,00**		**w.1**	1		
4	**phi**	0	0	0	0		w.1=1	="w.1="&w.1		
5	0,04									
6	=A8+$A$5	=A*COS(w*t+phi)	=A*COS(w*t+phi)	=A*COS(w*t+phi)	=A*COS(w*t+phi)	=SUMME(B9:E9)				
7	**t**	**c.1**	**c.2**	**c.3**	**c.4**	**Sum**				
8	0,00	0,23	0,75	0,63	0,72	2,33		Schreibe Formel in B8!		
9	**0,04**	**0,23**	0,75	0,63	0,71	**2,31**		Kopiere nach B8:E808!		
807	31,96	0,20	0,35	-0,04	-0,41	0,10				
808	32,00	0,19	0,29	-0,11	-0,50	-0,13				

Abb. 2.12 (T) Vier Kosinusfunktionen c_1 bis c_4 werden im Bereich B8:E808 (unterhalb Γ) berechnet, in dessen Zellen überall dieselbe Formel steht, und in Spalte F summiert. Die Formeln in den Spalten werden in Zeile 6 in schräger Ausrichtung wiedergegeben. Die Zeit t ist als Spaltenvektor A8:A808 (links von Γ) definiert. Die Amplituden A, die Kreisfrequenzen ω und die Nullphasen ϕ sind (oberhalb Γ) als Zeilenvektoren B2:E2, B3:E3 und B4:E4 definiert. Die Kreisfrequenzen sind ein Vielfaches der Grundfrequenz ω_1

	A	B	C	D	E	F	G	H	I	J
1		=w.1			=D3+dw				=Delta.w/3	
2	**A**	0,5	1,5	1,5	0,5			**Delta.w**	1,00	
3	**w**	**2,00**	2,33	2,67	**3,00**			**dw**	**0,33**	
4	**phi**	0	0	0	0			**w.1**	2	
5	0,04						w.1=2; Delta.w=1			

Abb. 2.13 (T) Parameter für eine Schwebung. Die Kreisfrequenzen ω haben gleichen Abstand $d\omega$ (in I3 festgelegt) zueinander. Die Amplituden sind, bis auf einen Faktor 2, Binomialkoeffizienten und können mit einem pascalschen Dreieck bestimmt werden. Die bestimmenden Parameter stehen in I2 (Frequenzumfang $\Delta\omega$, aus dem $d\omega$ berechnet wird) und I4 (niedrigste Frequenz ω_1)

einer Formel beschreiben müssen, die dann in den gesamten Berechnungsbereich (Z. B. B8:E808) für die vier Funktionen kopiert wird.

2.3.3 Schwebungen

Vier Frequenzen, aber nur zwei Steuerparameter
Die Parameter unseres Rechenmodells werden geändert, um Schwebungen zu erzeugen. Die Frequenzen sollen wiederum äquidistant sein. Sie haben denselben Abstand zueinander. Als Steuerparameter nehmen wir die niedrigste Frequenz ω_p und die Breite im Frequenzbereich (spektrale Breite $\Delta\ \omega$), also die Differenz zwischen der höchsten und der tiefsten Frequenz, siehe Abb. 2.13 (T).

Fragen

Wie bestimmt man aus der spektralen Breite $\Delta\omega$ den Abstand $d\omega$ zwischen den vier diskreten Frequenzen innerhalb des Bandes?[9]

▶ **Mag** Die Amplituden der Kosinuskomponenten einer Schwebung sind nicht beliebig, sondern Binomialkoeffizienten. Wie erhält man solche Frequenzen?

▶ **Tim** Nun, mit einem pascalschen Dreieck.

▶ **Alac** Warum so genaue Vorschriften für die Amplituden? Können wir die Amplituden nicht beliebig wählen?

▶ **Mag** Bei der genannten Wahl der Amplituden erhalten wir ein klares Bild im Zeitbereich, die Einhüllende ist ähnlich einer Glockenkurve. Sie sollen zunächst nur die Grundfrequenz und die spektrale Breite ändern und beobachten, wie sich die Schwebung im Zeitbereich verhält. Später können Sie die Amplituden beliebig wählen und herausfinden, ob die beobachtete Gesetzmäßigkeit erhalten bleibt.

[9] $d\omega = \Delta\omega/3$ (nicht $\Delta\omega/4$).



Abb. 2.14 **a** (links) Schwebung aus vier Kosinusfunktionen; Anfangskreisfrequenz $\omega_1 = 4$; Breite im Frequenzbereich $\Delta\omega = 3$. **b** (rechts) Schwebung aus vier Kosinusfunktionen; Anfangskreisfrequenz $\omega_1 = 2$; Breite im Frequenzbereich $\Delta\omega = 1$

Schwebungen entstehen, wenn man Kosinusfunktionen addiert, deren Frequenzen äquidistant sind. Das Signal ballt sich zu Paketen zusammen. In Abb. 2.14a und b werden die Pakete durch schwarze Punkte getrennt. Wir wollen die Breite solcher Wellenpakete bestimmen.

Aufgabe

Sie sollen Trennpunkte in die Diagramme einfügen, die in den Knoten liegen. Erzeugen Sie in der Tabelle Formeln, die die Koordinaten der Trennpunkte angeben, wenn die Breite eines Paketes Δt und eine anfängliche Zeitverschiebung t_0 vorgegeben werden. Sie sollen die Koordinaten so verändern, am besten mit einem Schieberegler, dass die Punkte im Diagramm genau im Knoten der Schwebung liegen. Gibt es einen Zusammenhang zwischen zeitlicher Breite Δt und Frequenzumfang $\Delta\omega$?

▶ **Mag** Haben Sie die Regel für die Breite eines Wellenpaketes herausgefunden?

▶ **Tim** Der zeitliche Abstand zwischen zwei Paketen ist proportional zum Kehrwert der spektralen Breite, $\Delta t = 3\pi/\Delta\omega$. Die Breite ist sicherlich kleiner, weil das Signal ja am Rande seines Gebietes gegen null geht.

▶ **Mag** Damit haben Sie eine Unschärfebeziehung gefunden: $\Delta f \cdot \Delta t = 3/2 > 1$.

▶ **Tim** Heisenbergs Unschärfebeziehung?

▶ **Mag** Ja, es ist die Vertauschungsbeziehung von Zeit und Energie, wenn man berücksichtigt, dass $E = h \cdot f$ gilt.

2.3.4 Summenformel für den Kosinus

▶ **Mag** Wie schreibt man *cos(x)* + *cos(y)* als Produkt von zwei Kreisfunktionen?

▶ **Alac** Muss man so etwas wissen? Das kann man doch in einer Formelsammlung nachschlagen.

▶ **Tim** Ich habe mir die Besenweisheit gemerkt:

▶ Ψ *Cos plus Cos gibt Mittelwert mal halbe Differenz.*

▶ **Mag** Das ist eine gute Eselsbrücke, wenn Sie daraus die vollständige Form rekonstruieren können:

$$\cos(x) + \cos(y) = 2 \cdot \cos\left(\frac{x+y}{2}\right)\cos\left(\frac{x-y}{2}\right) \tag{2.6}$$

▶ **Alac** So einen verrückten Spruch werde ich nicht vergessen. Der erste Kosinus nimmt den Mittelwert der Argumente und der zweite Kosinus die halbe Differenz der Argumente.

▶ **Mag** Genau, und wenn $x = 0$ und $y = 0$, dann muss $1 + 1 = 2$ herauskommen. Das erklärt den Vorfaktor auf der rechten Seite von Gl. 2.6. Wir können die Summenformel mit unserem Rechenmodell überprüfen, indem wir je zwei von unseren vier Frequenzen gleich machen und alle Amplituden zu 0,5 setzen wie Abb. 2.15a (T).

Das Ergebnis der Tabellenrechnung in Abb. 2.15a (T) sieht man in Abb. 2.16 als dicke graue Kurve.

Die feinen schwarzen Kurven entsprechen den Funktionen

$$2 \cdot \cos\left(\left(\omega_1 + \frac{\mathrm{d}\omega}{2}\right)\mathrm{t}\right) \text{ (Mittelfrequenz)}$$

und

$$2 \cdot \cos\left(\frac{\mathrm{d}\omega}{2}\mathrm{t}\right) \text{ (halbe Differenzfrequenz)}$$

die wir in Abb. 2.15b berechnet haben.



Abb. 2.15 **a** (links, T) Parametersatz für eine Summe von zwei Kosinusfunktionen; $c_1 = c_3$; $c_2 = c_4$; in H5 steht die Legende für Abb. 2.16. **b** (rechts, T) Formeln für die schnelle Schwingung und die Einhüllende gemäß der Summenformel; Zelle \$F\$8 enthält die Amplitude, hier $A = 2$. K4 und L5 enthalten die Legende für die Kurven in Abb. 2.16



Abb. 2.16 Dicke graue Kurve: Summe aus zwei Kosinusfunktionen mit den Frequenzen ω_1 und $\omega_1 + \mathrm{d}\omega$. Schwarze gepunktete Kurve: Kosinusfunktion mit der Mittelfrequenz; schwarze durchgezogene Kurve (Einhüllende): Kosinusfunktion mit der halben Differenzfrequenz, Tabellenorganisation dazu in Abb. 2.15 (T)

Wir sehen, dass der Kosinus mit der halben Differenzfrequenz, berechnet in Spalte L, die Summe einhüllt. Die schnellen Schwingungen, berechnet in Spalte K, haben die Mittelfrequenz. Das Produkt erfährt nach jedem Nulldurchgang der Einhüllenden eine Phasenverschiebung von π.

2.4 Polarkoordinaten (G)

Wenn wir einen Kreis darstellen wollen, dann verwenden wir am besten ebene Polarkoordinaten. Zur Festlegung eines Punktes in der Ebene gibt man dabei einen Winkel ϕ zur x-Achse und den Abstand r zum Nullpunkt an wie in Abb. 2.17a.



Abb. 2.17 **a** (links) Kartesische Koordinaten (x, y) und ebene Polarkoordinaten (r, φ). **b** (rechts) Kreis um den Nullpunkt und verschobener Kreis, Koordinaten aus Abb. 2.18 (T)

Polare und kartesische Koordinaten
Ebene Polarkoordinaten (r, φ) und kartesische Koordinaten (x, y) in der Ebene hängen folgendermaßen zusammen, siehe Abb. 2.17a:

$$x = r \cdot \cos(\varphi);\; y = r \cdot \sin(\varphi) \tag{2.7}$$

$$r = \sqrt{x^2 + y^2};\, \varphi = \arctan 2(y; x) \tag{2.8}$$

$$m = \tan\varphi = \frac{y}{x} \tag{2.9}$$

Dabei ist m die Steigung der Radiusstrecke.

Zur Berechnung des Winkels gegen die Horizontale aus den kartesischen Koordinaten eines Punktes verwendet man am besten die Tabellenfunktion ARCTAN2. Der Argumentbereich des klassischen *arcus tangens* reicht nur von $-\pi/2$ bis $\pi/2$, siehe Abschn. 5.10. Wenn man die Tabellenfunktion ARCTAN anwendet, dann muss mit den Vorzeichen von x und y zusätzlich geprüft werden, ob der Punkt in der Ebene $y \geq 0$ oder < 0 liegt. Die Tabellenfunktion ARCTAN2 führt diese Prüfung selbst durch.

2.5 Wellenfronten und machscher Kegel

Wir zeichnen die Wellenfronten einer Schallquelle, die sich mit veränderbarer Richtung und Geschwindigkeit in der xy-Ebene bewegt. Mit Einsatz eines Schiebereglers, der die Fluggeschwindigkeit bestimmt, durchbrechen wir die Schallmauer. In der Tabellenkalkulation wechseln wir zwischen polaren und kartesischen Koordinaten.



	A	B	C	D	E	F	G	H	I	J	K	L
1							=G3+dphi	=r.0*COS(phi.K)	=r.0*SIN(phi.K)	=x.K+v.x*t.0	=y.K+v.y*t.0	
2	**r.0**	1,20					**phi.K**	**x.K**	**y.K**	**x.s**	**y.s**	
3	**dphi**	0,17					0,00	1,20	0,00	-1,30	-1,65	
4	**v.x**	-0,41	9				**0,17**	1,18	0,21	-1,32	-1,44	
5	**v.y**	-0,27	23				0,35	1,13	0,41	-1,37	-1,24	
6	**t.0**	6,10	61				0,52	1,04	0,60	-1,46	-1,05	
39							6,28	1,20	0,00	-1,30	-1,65	
40												
41							v.x=-0,41; v.y=-0,27; t.0=6,1			0,00	-1,65	
42										-2,50	-1,65	
43										-2,50	0,00	

Abb. 2.18 (T) Koordinaten für einen Kreis um den Nullpunkt (G3:I39), für einen verschobenen Kreis (J3:K39) und für den Mittelpunkt des verschobenen Kreises (J42:K42); die Schieberegler bestimmen die Geschwindigkeit (v_x, v_y), mit der der Kreis verschoben wird und die Zeit t_0. B4 = [=(C4 − 50)/100]; B5 = [=(C5 − 50)/100]

Kreis, um den Nullpunkt und verschoben

In einer Tabelle werden die Punkte eines Kreises am besten zunächst in Polarkoordinaten mit konstantem Winkelabstand $d\varphi$ berechnet und dann in kartesische Koordinaten umgerechnet, so wie es in Abb. 2.18 (T) gemacht wird. Ein Kreis wird, wie jede Funktion, in ein Diagramm mit zwei Datenreihen eingetragen, eine für die x- und eine für die y-Werte.

In Spalte G werden 37 Winkel φ_K in gleichmäßigen Abständen $d\varphi = 2\pi/36$ (in B3) definiert. Daraus und aus r_0 in B2 werden dann die kartesischen Koordinaten (x_K, y_K) für einen Kreis um den Nullpunkt bestimmt, der in Abb. 2.17b grafisch mit Rauten für die Datenpunkte wiedergegeben wird.

Die Verschiebung des Kreises wird mit einem „Geschwindigkeitsvektor" (v_x, v_y) in B4:B5 und der „Zeit" t_0 in B6 von Abb. 2.18 (T) festgelegt. Alle drei Größen sollen mit Schiebereglern eingestellt werden. Die Koordinaten (x_s, y_s) des verschobenen Kreises stehen in den Spalten J und K. Wenn die Schieberegler für die Zellen C5 und C6 betätigt werden, mit denen v_x und v_y in B4 und B5 bestimmt werden, dann soll sich der Kreis waagrecht bzw. senkrecht verschieben. Wenn der Schieberegler für t_0 betätigt wird, dann verschiebt sich der Kreis auf einer Geraden durch den Nullpunkt. So soll es jedenfalls sein, wenn alle Formeln richtig eingesetzt wurden.

Fragen

Welche Funktion haben die Koordinaten J41:K43 relativ zu denen des Mittelpunkts des Kreises in J42:K42?[10]

Koordinaten der kreisförmigen Wellenfronten

Ein Flugobjekt bewegt sich parallel zur Erdoberfläche relativ zur Luft mit der Geschwindigkeit v_Q und unter dem Winkel α zur Horizontalen. Es sendet

[10] Die Koordinaten verbinden den Mittelpunkt des Kreises jeweils senkrecht mit den Achsen.

kontinuierlich Schallwellen aus, die sich mit der Schallgeschwindigkeit c_S ausbreiten. Es ist bei $t = 0$ bis zum Ort (0, 0) gekommen. Wir zeichnen jede zurückliegende Sekunde eine kreisförmige Wellenfront.

Der Mittelpunkt der Kreise wird durch

$$x_M(t) = \left(v_Q \cdot \cos(\alpha)\right) \cdot t \tag{2.10}$$

$$y_M(t) = \left(v_Q \cdot \sin(\alpha)\right) \cdot t \tag{2.11}$$

bestimmt. Die Ausdrücke in Klammern zerlegen die zurückgelegte Strecke $v_Q \cdot t$ in waagrechte (x) und senkrechte (y) Komponenten. $x_M(t)$ und $y_M(t)$ sind kleiner 0, weil in unserem Formelwerk $t < 0$ ist.

Von der Bahn des Flugobjekts ausgehend breiten sich Wellen aus, die bis zur Zeit $t = 0$ in jeder Richtung eine Strecke

$$r = -c_s \cdot t \text{ für } t \leq 0 \tag{2.12}$$

zurückgelegt haben, sodass eine (in unser Zeichenebene) kreisförmige Wellenfront entsteht. In der dreidimensionalen Wirklichkeit sind die Wellenfronten natürlich Kugelflächen. Die Größe r ist für unsere zweidimensionale Darstellung der Radius für einen Kreis in Abb. 2.19.

Berechnen von Kreisen

Wir berechnen in einer Tabelle acht Kreise, die die Wellenfronten für Wellen darstellen sollen, die zu den Zeitpunkten $t = -1$ s, -2 s,…, -8 s ausgesandt wurden.

Zwei Beispiele sehen Sie in Abb. 2.19a und b. Eine mögliche Tabellenorganisation dazu steht in Abb. 2.20 (T).

Die Definition der Kreiskoordinaten geschieht am besten in Polarkoordinaten über den Winkel φ, wie in Abschn. 2.4 beschrieben. In unserem Beispiel steht in



Abb. 2.19 **a** (links) Wellenfronten einer akustischen Welle, die von einer bewegten Schallquelle ausgesandt werden, die zur Zeit $t = 0$ bis (0,0) gekommen ist; Fluggeschwindigkeit gegen Luft v_Q (hier Unterschallgeschwindigkeit) und Winkel α gegen die Horizontale. **b** (rechts) Wie a, aber $\alpha = 10°$ und Überschallgeschwindigkeit

[A13:A43] $\varphi = 0$ bis 2π in 30 Schritten von $d\varphi = 0{,}209 = 2\pi/30$, definiert in A10. Wir wählen also den Winkel φ als unabhängige Variable.

Die Tabelle in Abb. 2.20 (T) hat einen typischen Γ-Aufbau. Der Berechnungsbereich unterhalb Γ umfasst B13:I43. Der spaltenbezogene Parametersatz x, r, x_M, y_M für die acht Funktionen steht in B7:I9 oberhalb von Γ, gesteuert von der Zeit t (in Zeile 6), siehe die Formeln in J7:J9. Die unabhängige Variable ϕ steht in A13:A43, links von Γ. Dieser Aufbau kann zeilenmäßig über die Zeile 43 vergrößert werden, weil er nach unten offen ist. Spaltenmäßig kann er nicht verbreitert werden, da sich ab Spalte K ein anderer Berechnungsbereich anschließt, in dem die y-Koordinaten der Kreise berechnet werden.

Die drei Parameter der Aufgabe (Schallgeschwindigkeit c_s, Fluggeschwindigkeit v_Q, Flugwinkel α) werden in B1:B4 festgelegt und mit den Namen in A1, A2, A4 benannt. Mithilfe dieser Parameter werden in den Zeilen 7 bis 9 die Radien und die Mittelpunkte der Kreise zu verschiedenen Zeitpunkten berechnet. Die Formeln in I7:I9 stehen rechts neben diesen Zellen, Z. B. I7 = [=−*c.S**I6]. Die Formeln in B7:H9 sehen entsprechend aus. Denken Sie daran:

	A	B	C	D	E	F	G	H	I	J	K	L	M
1	**c.S**	340 m/s		*Schallgeschwindigkeit*									
2	**v.Q**	500 m/s		*Geschw. der Quelle*				*="vQ="&vQ&"; alpha="&B3&"°"*					
3		10 °		*Winkel der Flugrichtung*				vQ=500; alpha=10°					
4	**alpha**	**0,175** rad		*=$B$3/180*PI()*									
5													
6	**t [s]**	-1	-2	-3	-4	-5	-6	-7	-8				
7	**r [m]**	340	680	1020	1360	1700	2040	2380	**2720**	*=-c.S*I6*			
8	**x.M [m]**	-492	-985	-1477	-1970	-2462	-2954	-3447	**-3939**	*=v.Q*COS(alpha)*I$6*			
9	**y.M [m]**	-87	-174	-260	-347	-434	-521	-608	**-695**	*=v.Q*SIN(alpha)*I$6*			
10	0,209	*=2*PI()/30*											
12	**phi**	**x.1**	**x.2**	**x.3**	**x.4**	**x.5**	**x.6**	**x.7**	**x.8**		**y.1**	**y.2**	**y.3**
13	0,000	**-152**	-305		**-610**	-762	-914	**-1067**	-1219		**-87**	-174	-260
14	**0,209**	-160	**-320**	-480	-639	**-799**	-959	-1119	**-1279**		-16	**-32**	-48
43	6,283	-152	-305	**-457**	-610	-762	**-914**	-1067	-1219		-87	-174	**-260**

Abb. 2.20 (T) Koordinaten von Kreisen, deren Mittelpunkte in der xy-Ebene verschoben werden (unterhalb Γ); die x-Koordinaten in B13:I43 sollen durch Kopieren einer Formel in Zelle B13 erzeugt werden, die mit absoluter und relativer Adressierung geeignet gestaltet werden muss; Winkelkoordinaten ϕ für alle Kurven in A13:A43 (unabhängige Variable links von Γ). Die Zeit in Zeile 6 steuert den Radius (in Zeile 7) und die Koordinaten des Mittelpunktes (in Zeile 8 und Zeile 9). Die Formeln für A13:I143 stehen in der ausgeblendeten Zeile 11, die in Abb. 2.21 (T) einzusehen ist. Die zugehörigen y-Koordinaten stehen in den Spalten K:R

	A	B	C	D	E	F	G	H	I	J	K	L	M
11	*=A13+$A$10*	*=B$7*COS($A13)+B$8*	*=C$7*COS($A14)+C$8*	*=D$7*COS($A43)+D$8*	*=E$7*COS($A13)+E$8*	*=F$7*COS($A14)+F$8*	*=G$7*COS($A43)+G$8*	*=H$7*COS($A13)+H$8*	*=I$7*COS($A14)+I$8*		*=B$7*SIN($A13)+B$9*	*=C$7*SIN($A14*	*=D$7**
12	**phi**	**x.1**	**x.2**	**x.3**	**x.4**	**x.5**	**x.6**	**x.7**	**x.8**		**y.1**	**y.2**	**y.3**
13	0,000	**271**	541		**1083**	1354	1624	**1895**	2166		**-40**	-80	-120
14	**0,209**	263	**527**	790	1053	**1316**	1580	1843	**2106**		31	**61**	92
43	6,283	271	541	**812**	1083	1354	**1624**	1895	2166		-40	-80	**-120**

Abb. 2.21 (T) Kopie eines Teils von Abb. 2.20 (T) mit eingeblendeten Formeln; die Formeln in Zeile 11 beziehen sich auf die Zellen mit den fett gedruckten Zahlen. Bevor Sie alle Formeln abschreiben, überlegen Sie lieber, wie sie zustande gekommen sind und entwickeln Sie sie dann aus Ihrem Verständnis heraus!

▶ Wenn Sie die Formel in einer Zelle richtig mit relativen und absoluten Bezügen geschrieben haben, dann können Sie sie ohne Veränderung in den gesamten Zellbereich ziehen.

Unterhalb des Γ werden in B13:I43 acht Kreise um den Nullpunkt mit dem Radius r aus Zeile 7 berechnet und um (x_M, y_M) aus den Zeilen 8 und 9 verschoben. Die Formeln in der ausgeblendeten Zeile 11, die für die fett gedruckten Zellen gelten, stehen in Abb. 2.21 (T).

Fragen

Fragen zu Abb. 2.20 (T):

Welche Bedeutung hat die Formel in Zelle B4? Der Formeltext steht in D4![11]

Welche Formel steht in Zelle D7?[12]

Welche Formel steht in Zelle D9?[13]

Wie können Sie mit einem Befehl alle acht Namen in A12:I12 vergeben?[14]

▶ **Tim** Die Wellenkämme in meinem Diagramm bleiben immer gleich, auch wenn ich die Fluggeschwindigkeit ändere.

▶ **Mag** Sie haben in ihr Tabellenblatt im Bereich B7:I9 die *Zahlen* aus Abb. 2.20 (T) eingeschrieben. Die genannten Zellen sollen aber *Formeln* enthalten, nicht nur Zahlen. Dann ändern sich die Werte in diesen Zellen, wenn die Parameter des Problems verändert werden. Die Formeln für die Spalte I finden Sie in Spalte J. Wenn Sie die Zellen I7:I9 bis in die Spalte B ziehen, dann ist das Formelwerk für die Parameter fertig.

▶ **Tim** Was ist genau unsere Aufgabe?

▶ **Mag** Die Winkel φ stehen in Spalte A. Sie sollen jetzt die Zellen B13 und K13 mit Formeln beschreiben, die durch Ziehen bis I43 bzw. R43 alle acht Kreise erzeugt. Überlegen Sie zunächst selbst und schauen Sie bei Bedarf in Abb. 2.21 (T) nach!

Die Formeln im Berechnungsbereich B13:I43 von Abb. 2.20 (T) bestehen aus zwei Termen, den Koordinaten eines Kreises um den Nullpunkt mit dem in Zeile 7

---

[11] Umrechnung des Winkels von Grad in Radiant: $360° = 2\pi$.

[12] [D7] = [=−c.S*D6].

[13] [D9] = [=v,Q*sin(alpha)*D6].

[14] Man markiert den Bereich A12:I43, ruft FORMELN/DEFINIERTE NAMEN/AUS AUSWAHL ERSTELLEN auf und bestätigt mit NAMEN ERSTELLEN AUS DEN WERTEN IN/OBERSTER ZEILE.

aus Gl. 2.12 berechneten Radius und einer Verschiebung gemäß der Geschwindigkeit v_Q der Quelle, Gl. 2.10 und 2.11.

Das Flugobjekt soll sich zum Zeitpunkt $t = 0$ bis zum Nullpunkt des Koordinatensystems (0, 0) bewegt haben. Die Zeit in Zeile 6 von Abb. 2.20 (T) wird rückwärts gezählt. Die Koordinaten (x_M; y_M) in den Zeilen 8 und 9 geben also an, wo sich das Flugobjekt vor 1, 2, usw. Sekunden befand. Sie ergeben sich, wie oben erläutert, aus der Geschwindigkeit der Quelle (v_Q), dem Winkel α, mit dem sich die Quelle gegen die x-Achse bewegt und dem Zeitpunkt t. Der jeweilige Ort des Flugobjektes bei dieser zurückliegenden Zeit ist dann auch der Mittelpunkt der Kreise der dann ausgesandten Schallwellen.

Schallmauer und machscher Kegel

▶ **Mag** Was passiert, wenn ein Flugzeug genau mit der Schallgeschwindigkeit fliegt?

▶ **Alac** Dann wird die Schallmauer durchbrochen und es gibt einen lauten Knall.

▶ **Mag** Simulieren Sie das! Am besten bauen Sie dazu einen Schieberegler ein und erhöhen die Geschwindigkeit der Schallquelle von $v_Q = 0$ ausgehend langsam bis zur Schallgeschwindigkeit!

▶ **Tim** Was heißt „Schallmauer durchbrechen“?

▶ **Mag** Alle Schallwellen treffen an einem bestimmten Ort (demjenigen der „Schallmauer“) zur selben Zeit ein und verstärken sich.

Aufgabe
Bauen Sie zwei Schieberegler ein, mit denen Richtung und Geschwindigkeit der Quelle eingestellt werden können und beobachten Sie, wie das Diagramm auf Veränderungen der beiden genannten Parameter und auf eine Veränderung der Schallgeschwindigkeit reagiert!

Ausgeblendete Formeln
B13 und K13 in Abb. 2.20 (T) werden mit einer Formel beschrieben, die dann in den restlichen Tabellenbereich durch Ziehen kopiert werden kann. Es gilt: B13 = [=B$7*cos($A13) + B$8] und K13 = [=B$7*sin($A13) + B$9].

2.6 Exponentialfunktionen

Für Exponentialfunktionen gilt: Plus 1 im Argument ergibt mal e im Wert. Man zeichnet eine Exponentialfunktion $A \cdot \exp(-t/t_0)$ von Hand (? Ja, auch von Hand!) am besten, indem man mit der Tangente bei $t = 0$ anfängt. Diodenkennlinien sind durch eine Knickspannung gekennzeichnet.

2.6.1 Reiskörner auf Schachbrett (G)

Wette im alten Orient

In einer klassischen orientalischen Wette vereinbarte ein listiger Schachspieler, dass sein Gewinn ausgezahlt werde, indem er auf das erste Feld eines Schachbretts ein Reiskorn legte und sein unterlegener Gegner dann auf alle folgenden Felder die doppelte Anzahl der jeweils vorhergehenden lege.

In Abb. 2.22a (T) wird die Vermehrung der Reiskörner nach diesem Verfahren nachgebildet.

In A3:A66 werden die 64 Felder des Schachbretts von 0 bis 63 durchnummeriert. In B3 wird ein Reiskorn auf das erste Feld ($n = 0$) gelegt. In den folgenden Zellen wird die vorhergehende Anzahl jeweils verdoppelt, bis in B66 ($n = 63$) die riesige Anzahl $2^{63} = 9{,}22 \times 10^{18}$ erreicht wird.

▶ Für $y = 2^x$ gilt: Ψ *Plus 1 (im Argument) wird zu mal 2 (im Wert).*

Die Werte werden in Abb. 2.22b grafisch dargestellt. Man sieht, dass die Explosion um den Faktor 10^{19} auf den letzten Feldern stattfindet.

Exponentialzahl in der Tabelle

Die Verdoppelung wird in der Tabelle immer weiter geführt, bis die entstehende Zahl nicht mehr dargestellt werden kann. Ab Zeile 66 in Abb. 2.22a (T) werden die Zahlen als Exponentialzahl wiedergegeben, 9,22E+18 $= 9{,}22 \times 10^{18}$. Die Zahl 2^{1024} ist in EXCEL nicht mehr darstellbar, siehe Zeile 1027.

	A	B	C	D	E
1	=A3+1	=B3*2	=2^A4		=2^D4
2	n	y	y =2^n		2^62,4
3	0	1	1		
4	1	2	2	62,4	6,1E+18
5	2	4	4		
6	3	8	8		
7	4	16	16		
8	5	32	32		
9	6	64	64		
66	63	9,22E+18	9,22E+18		
67	64	1,84E+19	1,84E+19		
1025	1022	4,5E+307	4,5E+307		
1026	1023	9,0E+307	9,0E+307		
1027	1024	#ZAHL!	#ZAHL!		



Abb. 2.22 **a** (links, T) Zweierpotenzen $y = 2^n$, durch fortlaufenden Multiplikation mit 2 (Spalte B) und durch Potenzierung (C:E). **b** (rechts) grafische Darstellung der Zweierpotenzen aus a

Gewinn

Fragen

Welchen Anteil am im Jahre 2003 international gehandelten Reis hätte der Gewinner bekommen, wenn der Verlierer (ein überaus reicher mittelalterlicher Sultan) hätte liefern können?[15]

Welchen Wert hat die Zahl mit der Darstellung 10E3?[16]

▶ **Mag** Kann der Gewinner von seinem Gewinn satt werden?

▶ **Mac** Vielleicht einmal von den Reiskörnern, die auf dem Schachbrett liegen.

▶ **Tim** Ich habe gehört, dass eine Exponentialfunktion explodieren kann. Also vielleicht kann der Gewinner eine Woche gut vom gewonnenen Reis leben.

▶ **Mag** Die Weltjahresproduktion von Reis lag 2015/16 bei 470 Mio.Tonnen, aber nur etwa 5 % kamen auf den Weltmarkt. Anders als Weizen wird Reis zu mehr als 95 % in den Anbauländern verzehrt.

▶ **Tim** Ich habe gezählt. Ein Kilogramm Reis enthält etwa $40.000 = 4 \times 104$ Körner. Auf dem Schachbrett sollte also nach der Spielregel mehr als das 10.000-fache des Welthandels von Reis liegen. Unglaublich!

▶ **Alac** Wahnsinnig! Eine Katastrophe! Das kann doch nicht stimmen. Wo ist der Haken?

▶ **Mag** Es gibt keinen Haken. Die Katastrophe kommt durch Ψ *Plus 1 wird zu mal 2* zustande.

In Spalte C von Abb. 2.22a (T) wird die Potenzfunktion $y = 2^n$ berechnet. Sie wird als Tabellenformel [=2^n], bzw. im konkreten Beispiel C4 = [=2^A4] eingegeben und ergibt dieselben Ergebnisse wie die Multiplikationen mit 2 in Spalte B. Das Argument der Potenzfunktion muss keine ganze Zahl sein. In E4 wird der Wert für das Argument 62,4 in D4 berechnet und als gefüllte Raute in Abb. 2.22b eingefügt.

Exponentialfunktion
Die Potenzfunktion lässt sich verallgemeinern zu $y = a^x$, wobei x eine reelle Zahl ist. Wenn a die eulersche Zahl $e = 2{,}718$ ist, dann wird die Potenzfunktion zur bekannten Exponentialfunktion, deren Tabellenformel [=EXP(…)] lautet.

[15] Bei 64 Feldern hätte der Gewinner $2^{64}-1 = 18 \times 10^{18}$ Körner entsprechend etwa 10^{18} g $= 10^{12}$ Tonnen Reis bekommen. Das ist die 2000-fache Menge der Reisernte von 2015/16 von 470 Mio. Tonnen.

[16] $10E3 = 10 \times 10^3 = 10^4 = 10.000$.

2.6.2 Grafische Darstellung (G)

Exponentialfunktion
Die Exponentialfunktion wird in mathematischen Lehrbüchern meist folgendermaßen geschrieben:

$$f(x) = A \cdot \exp(ax) = A \cdot e^{ax}$$

Es ist jedoch physikalisch gesehen einsichtiger, sie so zu schreiben:

$$f(x) = A \cdot \exp\left(\frac{x}{\mathrm{x}_0}\right) = A \cdot e^{\frac{x}{x_0}} \tag{2.13}$$

Damit ist die Einheit von x_0 gleich der Einheit von x, also z. B. eine Länge oder eine Zeit und hat eine anschauliche Bedeutung: Die Tangente bei $x = 0$ schneidet die y-Achse bei der Amplitude A und die x-Achse bei der charakteristischen Länge x_0.

Der Formelbuchstabe e bezeichnet die eulersche Zahl $e = \sum_0^\infty \frac{1}{n!} = 2,718$.

Anfangssteigung der Exponentialfunktion
In Abb. 2.23 (T) werden zwei Exponentialfunktionen

$$f_e(t) = A_e \exp\left(\frac{t}{t_e}\right) = A_e \cdot e^{\frac{t}{t_e}} \tag{2.14}$$

für 51 äquidistante Stützstellen berechnet, deren Parameter A_e und t_e in B1:D1 bzw. B2:D2 als benannte Zeilenbereiche stehen. Die charakteristische Zeit t_e ist einmal positiv und einmal negativ. Der Abstand dt der Stützstellen wird in G1 festgelegt.

▶ Für die Exponentialfunktion $y = e^x$ gilt Ψ *Plus 1 wird zu mal e.*
Für die Zweierpotenz $y = 2^n$ gilt Ψ *Plus 1 wird zu mal 2.*

Die Abb. 2.23 (T) hat einen typischen Γ-Aufbau Die Zeit t ist die unabhängige Variable. Ihre 51 Werte t_i stehen im mit „t" benannten Spaltenbereich A7:A57 links von Γ. Die Parameter A_e und t_e der Kurven stehen zeilenweise oberhalb vom Γ. Das Zeitintervall dt, das heißt der Abstand zwischen den Stützstellen auf der waagrechten Achse, wird in Zelle G1 festgelegt. Der Anfangswert, hier −20, wird in Zelle A8 geschrieben. Die Werte für die restlichen 50 t_i-Werte werden nacheinander aus dem jeweiligen Vorgänger ermittelt. Der t-Vorgänger für Zelle A8 ist die Zelle A7, A8 = [=A7 + dt]. In der Formel ist A7 nicht mit Dollarzeichen versehen. Es handelt sich also um einen relativen Bezug, sodass beim Kopieren die angesprochene Zelladresse verändert wird. Die Formel in A57 lautet somit A57 = [=A56 + dt].

	A	B	C	D	E	F	G	H
1	**A.e**	3,0		1,0		**dt**	2,0	
2	**t.e**	-30,0		15,0				
4	*=A7+dt*	*=A.e*EXP(t/t.e)*	*=A.e&"·exp(t/"&t.e&")"*	*=A.e*EXP(t/t.e)*		*=-B2*	*=B1*	
5		3·exp(t/-30)		1·exp(t/15)				
6	**t**	**exp.1**		**exp.2**				
7	-20,0	5,8		0,3		0	**3,0**	*=B1*
8	**-18,0**	**5,5**		**0,3**		**30,0**	0	*=-B2*
9	-16,0	5,1		0,3				
10	-14,0	4,8		0,4		0,0	1,0	
11	-12,0	4,5		0,4		-15,0	0	
57	80,0	0,2		207,1				

Abb. 2.23 (T) Berechnung von zwei Exponentialfunktionen, deren unabhängige Variable t als benannter Spaltenbereich (in A7:A57) und deren Parameter Amplitude A_e und Zeitkonstante t_e als benannte Zeilenbereiche (in B1:D1 bzw. B2:D2) abgelegt werden. Die Legenden für die Funktionen werden in Zeile 4 mit der Formel in C4 zusammengestellt; Ψ *„Text" & Variablen*

Fragen

Welche Formeln stehen in F11 und G10 von Abb. 2.23 (T)?[17]

Welche Werte für die Zeit t stehen in Zelle A8 und A9, wenn A7 die Anfangszeit $t = 5$ enthält und die Länge eines Zeitabschnitts in einer Zelle mit Namen dt gespeichert ist und den Wert 2 hat?[18]

Wie können die Koordinaten der Tangenten bei $t = 0$ aus den Parametern der Funktionen analytisch bestimmt werden?[19]

Wie groß ist der Abstand der Stützstellen der Funktionen in Abb. 2.24a?[20]

Zuerst die Tangente bei $t = 0$!
Im Bereich F7:G11 von Abb. 2.23 (T) werden die Koordinaten für die Tangenten an die Exponentialfunktionen bei $t = 0$ berechnet.

Aufgabe
Stellen Sie die beiden Exponentialfunktionen zusammen mit ihren Tangenten bei $t = 0$ in einem Diagramm dar! Zwei Beispiele sehen Sie in Abb. 2.24a.

Aufgabe
Verändern Sie die Parameter A_e, t_e und dt und überprüfen Sie, ob die Abbildung entsprechend reagiert! Das tut sie, wenn jede Zelle die richtige Formel enthält. Erinnern Sie sich: Sie können nicht einfach die Zahlen aus den Abbildungen in

[17] F11 = [=−D2]; G10 = [=D1]; Koordinaten der Tangente bei $t = 0$.

[18] A8 = [=A7 + dt] = 7; A9 = [=A8 + dt] = 9.

[19] Gerade durch die beiden Punkte (0,A) und (t_e,0).

[20] Der Abstand der Stützstellen ist dt = 2, siehe Abb. 2.23 (T), G1.



Abb. 2.24 **a** (links) Die zwei Exponentialfunktionen der Abb. 2.23 (T) mit ihren Tangenten an den Schnittpunkten mit der y-Achse. **b** (rechts) So zeichnet man eine Exponentialfunktion von Hand: Zuerst die Gerade mit den Schnittpunkten auf der *x*- und *y*-Achse!

Ihre Tabellenblätter übertragen. Die meisten Zellen enthalten eine Formel, nur in manche wird direkt eine Zahl eingetragen.

▶ **Mag** Wissen Sie jetzt, wie man eine Exponentialfunktion von Hand auf einem Blatt Papier zeichnet?

▶ **Alac** Klar! Zuerst die Punkte A_e auf der y-Achse und t_e auf der x-Achse markieren und eine Gerade hindurchlegen. Die exponentielle Kurve schmiegt sich dann bei $t = 0$ an die Tangente und für $t \to \infty$ oder $t \to -\infty$ an die x-Achse an.

▶ **Mag** Richtig, wie Abb. 2.24b! Merken Sie sich:

▶ Ψ *Expo mit Knick und Gerade*

▶ **Tim** Gerade ist klar, aber wieso Knick?

▶ **Mag** Das werden Sie gleich in Abschn. 2.6.3 sehen.

2.6.3 Diodenkennlinie bei verschiedener Skalierung der *I*-Achse

Der Strom I durch eine Halbleiterdiode hängt exponentiell von der angelegten Spannung U ab. Die $I = I(U)$-Kennlinie einer Halbleiterdiode wird durch eine Exponentialfunktion beschrieben, die durch Null geht:

$$I = I_s \cdot \left(\exp\left(\frac{U}{U_T}\right) - 1\right) \tag{2.15}$$



	A	B	C	D
1	**I.s**		1,00E-14	
2	**U.T**		2,50E-02	
3	**dU**		2,50E-02	
4				
5				
6	=A8+dU	=I.s*EXP(U/U.T)	=I.s*(EXP(U/U.T)-1)	
7	**U**	**exp**	**I**	
8	-0,200	3,35E-18	-1,00E-14	
9	**-0,175**	9,12E-18	-9,99E-15	
58	1,050	1,74E+04	1,74E+04	

Abb. 2.25 **a** (links) Tabellenaufbau für Diodenkennlinie, I_s = Sättigungsstrom, U_T = Temperaturspannung. **b** (rechts) Diodenkennlinie, logarithmische Skalierung der I-Achse (semilogarithmische Darstellung). Die Kurve kann nur für $I > 0$ dargestellt werden, weil nur dafür der Logarithmus definiert ist

Diese Funktion hat zwei Parameter: die Stärke des Sperrstromes I_s und die Temperaturspannung U_T, die durch $k_B T/e$ gegeben ist, mit k_B = Boltzmannkonstante, e = Elementarladung und T = absolute Temperatur. Bei Zimmertemperatur ist $U_T = 25$ mV.

Wir wollen eine solche Funktion für $I_s = 1 \times 10^{-14}$ A und $U_T = 0{,}025$ V in verschiedenen Diagrammen darstellen. Ein mögliches Rechenmodell wird in Abb. 2.25a (T) wiedergegeben.

Die Parameter I_s und U_T der Diodenkennlinie werden in den benannten Zellen C1:C2 definiert. In C3 wird der Abstand dU zwischen benachbarten Punkten festgelegt. Wir haben hier d$U = U_T = 0{,}025$ V gewählt. Die I-U-Kennlinie wird in Abb. 2.25b im halblogarithmischen Maßstab dargestellt. In Abb. 2.26a wird sie im linearen Maßstab auch für negative Werte gezeigt. Sie schneidet die I-Achse bei 0. Es fließt also kein Strom, wenn keine Spannung anliegt.

Logarithmisch skalierte Achse

Die Achse eines Diagramms wird skaliert, indem man sie mit der linken Maustaste aktiviert und dann „Format" aufruft. In dem Fenster, das sich dann öffnet, lassen sich Minimum, Maximum und andere Parameter der Achse einstellen. Wenn die Achse logarithmisch skaliert werden soll, dann muss das entsprechende Kästchen aktiviert werden.

Fragen

Welche der Kurven in Abb. 2.26a und b sind Exponentialfunktionen? Welche sind auf der I-Achse verschoben? Welche sind auf der U-Achse verschoben?[21]

[21] Alle Kurven stellen Exponentialfunktionen dar. Die Diodenkennlinie in Abb. 2.26a ist auf der I-Achse um den Sättigungsstrom I_s nach unten verschoben, sodass sie durch den Nullpunkt geht. Die Kurven in Abb. 2.26b sind *nicht auf der U-Achse* verschoben worden.



Abb. 2.26 **a** (links) Diodenkennlinie und zugehörige Exponentialfunktion, Darstellung für kleine Ströme. **b** (rechts) Zweimal dieselbe Diodenkennlinie wie in a, aber mit anderer Skalierung der *I*-Achsen; linke *y*-Achse für die linke Funktion, rechte *y*-Achse für die rechte Funktion; die senkrechten Gitternetzlinien haben den Abstand $U_T = 25$ mV

> Bei welchen *U*-Werten in Abb. 2.26b liegen wohl die „Knickpunkte", von denen Elektroniker reden?[22]

Ebenfalls in Abb. 2.26a dargestellt ist die Exponentialfunktion ohne den Term -1 in der Klammer der Gl. 2.15. Diese Funktion schneidet die *I*-Achse bei I_s. Sie wächst um den Faktor *exp(1)* = 2.7, jedes Mal, wenn *U* um d*U* voranschreitet. Das Maximum der *I*-Achse liegt bei 9×10^{-14} A. Die Kurven in Abb. 2.26a sehen so aus wie man sich gewöhnlich eine Exponentialfunktion vorstellt.

Für Exponentialfunktionen zur Basis *e* gilt die Besenregel: Ψ *Plus 1 wird zu mal e*. Im konkreten Fall heißt das, wenn *U* um U_T (Abstand der vertikalen Gitternetzlinien in Abb. 2.26) voranschreitet, dann wächst *exp* um den Faktor 2,7.

In Abb. 2.26b wird dieselbe Kennlinie (derselbe Datensatz) wie in Abb. 2.26a zweimal dargestellt, lediglich mit einer anderen Skalierung der *I*-Achse; das *I*-Maximum ist jetzt bei $2{,}0 \times 10^{-5}$ A (linke vertikale Achse) bzw. bei $2{,}0 \times 10^{-3}$ A (rechte vertikale Achse). Die Kennlinien zeigen sich mit einem Knick bei etwa 0,5 V bzw. 0,65 V. In der Elektrotechnik wird diese Spannung als „Knickspannung" bezeichnet.

Der *U*-Wert steigt bei Fortschritt von $U = 0{,}475$ auf $U = 0{,}500$ (bzw. von 0,625 auf 0,65) um denselben Faktor 2,7 wie beim Fortschritt von $U = 0$ auf $U = 0{,}025$. Die Lage des offensichtlichen Knicks auf der *U*-Achse ist eine Funktion der Skalierung der *I*-Achse. Wenn die *U*-Werte von 0 zu 0,025 zu 0,05 fortschreiten, dann erscheint der Anstieg des Stromes *I* sanft wie in Abb. 2.24a. Wenn sie von 0,475 zu 0,500 zu 0,525 fortschreiten, dann erscheint der Anstieg des Stromes steil wie in Abb. 2.26b.

[22] Bei etwa 0,5 V und 0,6 V.

In Abb. 2.25b wird die U-Achse logarithmisch skaliert und die Kennlinie für $U > 0$ abgebildet. Die Kennlinie erscheint in dieser Darstellung als Gerade, bis auf die Punkte unterhalb $U = 0{,}05$, wo der Term -1 ins Gewicht fällt. Ein Knick ist nirgends zu erkennen.

2.7 Radioaktiver Zerfall

In dieser Übung berechnen wir Restmenge, Aktivität und mittlere Lebensdauer beim radioaktiven Zerfall. Dabei lernen wir:

- die Ergebnisse von Differenziation und Integration über dem richtigen x-Wert aufzutragen,
- den Mittelwert einer Funktion durch Integration zu bilden,
- Fallen bei der Bildung eines gewichteten Mittelwertes zu vermeiden.

2.7.1 Restmenge und Aktivität

Abklingzeit ≠ Halbwertszeit
Radioaktive Substanzen zerfallen spontan. Die Restmenge der noch nicht zerfallenen Substanz folgt einer Exponentialfunktion mit der Abklingzeit t_0:

$$N(t) = N_0 \cdot \exp\left(-\frac{t}{t_0}\right) \tag{2.16}$$

Die Abklingzeit ist nicht die Halbwertszeit. Die Halbwertszeit ist die Zeit, bei der die Hälfte der Anfangsmenge zerfallen ist.

Fragen

Die Halbwertszeit einer radioaktiven Substanz sei 10 Jahre. Ist die Abklingzeit kleiner oder größer? Um welchen Faktor?[23]

Wir untersuchen nun den Zerfall von zwei radioaktiven Substanzen N1 und N2 mit unterschiedlichen Halbwertszeiten $t_{½1}$ und $t_{½2}$. Die Parameter der Übung werden in Abb. 2.27 (T) gelistet

Die Anfangsmengen werden zu $N_{01} = 100$ und $N_{02} = 100$ gesetzt. Die Halbwertszeiten $t_{½1}$ und $t_{½2}$ betragen in diesem Beispiel 10 h bzw. 40 h. Sie wurden in die Zellen C2 und F2 eingetragen. In den Zellen C3 und F3 werden die Abklingzeiten t_{01}

[23]Die Abklingzeit ist um den Faktor $1/\ln(2) = 1{,}44$ größer als die Halbwertszeit.

	A	B	C	D	E	F	G	H	I
1	Anfangsmenge	**N.01**	100,00		**N.02**	100,00			
2	Halbwertszeit	**t(1/2).1**	10,00 h		**t(1/2).2**	40,00 h			
3	Abklingzeit	**t.01**	14,43 **h**		**t.02**	**57,7** =F2/LN(2)			
4	Zeitintervall	**dt**	5 ?						
5		N.01=100; t(1/2)=14,43			*="N.01="&N.01&"; t(1/2)="&RUNDEN(t.01;2)*				

Abb. 2.27 (T) Parameter von zwei Exponentialfunktionen, die radioaktiven Zerfall beschreiben sollen; die Legende in B5 kommt durch die Formel in E5 zustande. Die Abklingzeiten in C3 und F3 werden aus den Halbwertszeiten berechnet. Diese Tabelle wird in Abb. 2.28 (T) und 2.30 (T) fortgesetzt

	A	B	C	D	E	F	G	H	I	J	K
11	=A13+dt	=N.01*EXP(-t/t.01)	=(B13-N.1)/(t-A13)	=MITTELWERT(A13:A14)	=Aktivität_1*t.akt	=N.01/t.01*EXP(-(t+A13)/2/t.01)		=N.02*EXP(-t/t.02)	=(H13-H14)/($A14-$A13)	=Aktivität_2*t.akt	
12	t	N.1	Aktivität_1	t.akt	Lebensdauer 1	Aktivität 1 theoretisch		N.2	Aktivität_2	Lebensdauer 2	
13	0,00	100,00				6,93		100,00			
14	**5,00**	**70,71**	**5,86**	**2,50**	**14,64**	**5,83**		**91,70**	**1,66**	**4,15**	
33	100,00	0,10	0,01	97,50	0,79	0,01		17,68	0,32	32,00	
113	500,00	0,00	0,00	497,50	0,00	0,00		0,02	0,00	0,16	

Abb. 2.28 (T) Berechnung der Aktivität und der Lebensdauer von $t = 0$ bis 100 (Reihe 33) oder bis $t = 500$ (Reihe 113); N_1 und N_2 sind Exponentialfunktionen mit in Abb. 2.27 (T) vorgegebenen Halbwertszeiten; die Aktivitäten (Zerfälle pro Zeiteinheit) werden in Spalte C aus den Funktionswerten in Spalte B durch *numerische Differenziation* berechnet. Der Integrand zur Bestimmung der Lebensdauer nach Gl. 2.19 wird in der Spalte E gebildet

und t_{02} aus den Halbwertszeiten ausgerechnet. Beachten Sie, dass t_0 nicht die Halbwertszeit ist! Es gilt:

$$t_0 = \frac{t_{\frac{1}{2}}}{ln2} \tag{2.17}$$

Ein mögliches Rechenmodell zur Berechnung der Restmengen, der Aktivitäten und der Lebensdauern wird in Abb. 2.28 (T) dargestellt. Wenn Sie schon sicher mit Zellbezügen umgehen können, dann können Sie noch mehr Zellbezüge durch Namen ersetzen und so die Rechnung übersichtlicher machen.

Restmenge

Wie in Gl. 2.16 ist $N_i(t)$ die Restmenge (noch nicht zerfallene Substanz) zur Zeit t. N_{0i} ist die Menge zur Zeit $t = 0$ und t_{i0} ist die Abklingzeit (nicht die Halbwertszeit!) für die Substanz i, $i = 1$ oder 2, und wird gemäß Gl. 2.17 berechnet. In Abb. 2.28 (T) wird die Restmenge N_1 der Substanz 1 in der Spalte B berechnet, diejenige für N_2 in Spalte H. Die Kurven für die oben gewählten Parameter sehen Sie in der Abb. 2.29a.

Aufgabe

Stellen Sie die Restmenge der Substanzen als Exponentialfunktionen (Gl. 2.16) dar!



Abb. 2.29 **a** (links) Restmenge von zwei radioaktiven Substanzen mit kurzer und langer Halbwertszeit. **b** (rechts) Aktivität der Substanzen aus a; beachten Sie die Zeitpunkte, über denen die Aktivität aufgetragen ist!

Aktivität
Die Aktivität einer radioaktiven Substanz ist die Zahl der Zerfälle pro Zeiteinheit. Für unsere Rechnung wählen wir als Zeiteinheit 5 h. Die numerisch berechneten Aktivitäten für N_1 und N_2 finden Sie in den Spalten C bzw. I.

Aufgabe
Bestimmen Sie die Aktivität durch numerische Differenzbildung zwischen benachbarten Punkten der Kurve für die Restmenge! Überlegen Sie, über welchen Zeitpunkten Sie die ermittelten Werte am besten auftragen! Möglichkeiten: Anfang, Ende oder Mitte des Intervalls. Für welche Wahl ergibt sich die größte Übereinstimmung zwischen numerisch bestimmter und analytisch berechneter Aktivität? Holen Sie sich bei Bedarf Rat im Abschn. 5.2 „Differenzieren (G)“.

Aufgabe
Berechnen Sie die Anzahl der Zerfälle im Intervall dt durch formale Ableitung von Gl. 2.16 in einer Spalte der Tabelle (Spalte F in Abb. 2.28 (T)) und stellen Sie sie in einem Diagramm wie in Abb. 2.29b dar! Diese Größe nennt man die Aktivität des radioaktiven Materials. Achten Sie darauf, dass sie zu denselben Zeitpunkten berechnet wird, die Sie für die numerische Ableitung in Spalte C gewählt haben!

Fragen

Wie groß sind die Halbwerts- und Abklingzeiten der Zerfallskurven in Abb. 2.29a?[24]

[24]Für N01: Halbwertszeit 40 h, Abklingzeit ≈ 57 h; für N02: Halbwertszeit 10 h, Abklingzeit ≈ 15 h.

Die Restmenge einer radioaktiven Substanz ist 130 bei $t = 10$ s und 110 bei $t = 12$ s. Wie groß ist die Aktivität in diesem Zeitabschnitt? Über welchem Zeitpunkt trägt man diese Aktivität in einem Diagramm auf?[25]

2.7.2 Mittlere Lebensdauer durch gewichteten Mittelwert

Den Atomkernen, die im Zeitabschnitt zerfallen, der mit t_i anfängt und bei $t_i + dt$ endet, schreibt man die Lebensdauer

$$\tau(t) = t_i + \frac{dt}{2}$$

relativ zum gewählten Nullpunkt der Zeit zu. Die Zerfallsrate sei $n(t)$. Die mittlere Lebensdauer wird im Allgemeinen als Integral über den Zeitraum von 0 bis ∞ berechnet:

$$\langle\tau\rangle = \frac{\int_0^\infty \tau(t) \cdot n(t) dt}{\int_0^\infty n(t) dt} \tag{2.18}$$

Theoretisch ist die mittlere Lebensdauer gleich der Abklingzeit t_0.

Wir wollen die mittlere Lebensdauer numerisch aus unseren Datenpunkten berechnen, indem wir statt zu integrieren über alle Zeitintervalle aufsummieren:

$$\langle\tau\rangle = \frac{\sum_i \tau_i \cdot n_i}{\sum_i n_i} \tag{2.19}$$

Dabei ist n_i die Anzahl der Zerfälle im Zeitintervall *i*. Die mittlere Lebensdauer wird in Gl. 2.19 berechnet als *gewichteter Mittelwert* der Zeiten τ_i in der Mitte von Zeitintervallen gewichtet mit der Anzahl n_i der Zerfälle im Intervall als Gewichte. Weitere Informationen zur Bildung von gewichteten Mittelwerten finden Sie in Abschn. 5.7.

Aufgabe
Berechnen Sie numerisch die mittlere Lebensdauer eines Atomkerns aus den Daten in den Tabellen, Abb. 2.27 (T) und Abb. 2.28 (T) durch Integration über den Zeitbereich 0 bis 100 h (Reihen 13 bis 33)! Eine Beispielrechnung finden Sie in Abb. 2.30 (T).

Die Summanden des Zählers in Gl. 2.19 werden in Abb. 2.31 dargestellt, in a für $t = 0$ bis 100 und in b für $t = 0$ bis 500.

Für die Kurve „t.01=…“, dargestellt in Abb. 2.31a Daten aus den Spalten D und E von Abb. 2.28 (T), erhalten wir eine mittlere Lebensdauer von 14,5 h

[25] Die Aktivität ist 10/s. Sie wird über der Mitte des Intervalls, hier also bei $t = 11$ s aufgetragen. Siehe auch Abb. 2.29b für $dt = 5$ h.

	B	C	D	E	F	G	H	I	J	K	L
7		Summe Aktivität		mittlere Lebensdauer t.av				=SUMME(I14:I33)*dt			
8		**99,90**		**14,5**	*=E9/C8*dt*			**82,32**	**38,5**	*=J9/I8*dt*	
9	*=SUMME(C14:C33)*dt*			**289,2**	*=SUMME(E14:E33)*				**634,18**	*=SUMME(J14:J33)*	
10				t.01=14,4; t.av=14,6			t.02=57,7; t.av=38,5				

Abb. 2.30 (T) Berechnung der mittleren Lebensdauern t_{av} in E8 und J8 aus den Daten in Abb. 2.28 (T), Reihen 14 bis 33, t = 0 bis 100. Dieser Tabellenausschnitt steht zwischen Abb. 2.27 (T) und Abb. 2.28 (T)

(in Zelle E8 von Abb. 2.30 (T)). Das stimmt gut mit dem theoretischen Wert $t_{01} = 14{,}43$ h überein.

Fragen

Deuten Sie die Formeln in Zeilen 8 und 9 der Abb. 2.30 (T)![26]

Sie können die mittlere Lebensdauer mit der Formel E9 = [=Summenprodukt(*t.act*, *activity*)/SUMME(*Activity*)] berechnen. Raten Sie, was die Tabellenfunktion SUMMENPRODUKT macht![27]

Warum weist die Kurve „t.01" in Abb. 2.31a ein Maximum auf?[28]

Eine falsche (jedenfalls unvollständige) Tabellenrechnung

▶ **Tim** Für die zweite Kurve („t.02 = 57,7", große Halbwertszeit) in Abb. 2.31a ist die aus der Tabelle ermittelte mittlere Lebensdauer mit 38,5 h (J8 in Abb. 2.30 (T)) nur etwa 2/3 so groß wie die theoretische, $t_{02} = 57{,}7$ h (F3 in Abb. 2.27 (T)). Warum?

▶ **Mag** Weil Sie mit der Anweisung „Integration von 0 bis 100 h" in eine Falle gelockt worden sind! Sehen Sie sich die Summanden zur Berechnung der mittleren Lebensdauer in Abb. 2.31a an!

Aufgabe
Verändern Sie die Abklingzeiten und die Amplituden und finden Sie heraus, wann die numerisch ermittelte mittlere Lebensdauer der theoretischen entspricht! Hinweis: Betrachten Sie noch einmal Abb. 2.31b.

[26] In C8 wird die Aktivität aufsummiert. Wenn die Substanz vollständig zerfallen ist, dann muss die Anfangsmenge herauskommen. In E8 wird eine gewichtete Summe über die momentanen Lebensdauern gebildet, die in E14:E33 berechnet werden. Als momentane Lebensdauer in einem Intervall wird aus Spalte D (t_{akt}) der Mittelwert der Intervallgrenzen genommen. Das Gewicht ist die Aktivität in Spalte C.

[27] Das SUMMENPRODUKT multipliziert die Koeffizienten in den angegebenen Bereichen gleicher Größe und summiert die Produkte auf. Siehe dazu auch Abschn. 5.9.

[28] Bei kleinen Zeiten gibt es kleine Lebensdauern und große Aktivitäten, bei großen Zeiten gibt es große Lebensdauern und kleine Aktivitäten. Dazwischen kommt es zu einem Maximum.



Abb. 2.31 **a** (links) Lebensdauer t_i mal Zahl n_i der Zerfälle im Intervall i als Funktion der Zeit (Mitte des Intervalls i) zur Berechnung der mittleren Lebensdauern, Abb. 2.28 (T) Reihen 13 bis 33. **b** (rechts) Wie a, aber auf einer größeren Zeitskala, Reihen 13 bis 113

Aufgabe
Erweitern Sie für die Parameter der Abb. 2.27 (T) die Rechnung bis zur Zeit $t = 500$ h wie in Abb. 2.31b und vergleichen Sie die Abklingzeit mit der berechneten mittleren Lebensdauer!

In Abb. 2.31a werden der Inhalt von E14:E33 von Abb. 2.28 (T), das heißt der jeweilige Term $Aktivität_1 \cdot t_1$, und die entsprechenden Werte für die Substanz 2 als Funktion der Zeit bis $t = 100$ h dargestellt. Die mittlere Lebensdauer ist das normierte Integral über diese Kurven gemäß Gl. 2.18. In dieser Gleichung wird das Integral von 0 bis ∞ ausgeführt. „Bis ∞" lässt sich numerisch nicht machen. Wir können nur eine endliche Summe bilden. Diese Summe gibt nur dann die tatsächliche Lebensdauer wieder, wenn der Integrand im betrachteten endlichen Gebiet hinreichend schnell gegen null konvergiert. Das ist in Abb. 2.31a für t_{02} nicht der Fall, wohl aber in Abb. 2.31b, für die die Integranden bis $t = 500$, Reihe 113, dargestellt wird. Die Werte der Integrale stehen in der Legende.

▶ **Alac** Wenn Sie uns das gleich gesagt hätten, dann hätten wir uns die Doppelarbeit sparen können.

▶ **Mag** Sehen Sie, Sie haben nicht nur ein Rezept für die Mittelwertbildung angewandt, sondern auch noch praktische Erfahrung mit numerischen Fallstricken gesammelt. So bilden sie sich wissenschaftlich weiter :-)

2.8 Benennung von Zellbereichen: Konstanten, Vektoren, Matrizen

Wir definieren Spaltenvektoren, Zeilenvektoren und Bereichsmatrizen und lernen, wie man darauf komponentenweise und als Ganzes zugreift.

Der Namens-Manager

Wir wollen in einer Matrix die Werte $r = \sqrt{(x - x_1)^2 + y^2}$ berechnen, für x und y von jeweils -2 bis 2. Dazu erstellen wir ein Tabellenblatt wie in Abb. 2.32 (T).

Die x-Werte stehen in B3:F3, die y-Werte in A4:A8. Den Wert für x_1 haben wir in einem anderen Tabellenblatt namens „E.x“ definiert. Die Rechnung lässt sich mit gemischten Zellbezügen wie in Zelle B5 durchführen:

$$\text{B5} = \left[= \text{WURZEL}\left(\left(\text{B\$3} - \text{E.x!\$A\$1}\right)\hat{}2 + \text{A\$5}\hat{}2\right)\right], \tag{2.20}$$

wobei „E.x!$A$1“ auf die Zelle A1 im Tabellenblatt „E.x“ zugreift. Die Formel wird übersichtlicher, wie in Zelle D6 gemacht und in D2 gezeigt, wenn wir die Variablen mit Namen versehen:

$$\text{D6} = \left[= \text{WURZEL}\left((\text{x}-\text{x.1})\hat{}2 + \text{y}\hat{}2\right)\right] \tag{2.21}$$

Dazu müssen wir z. B. den Bereich B3:F3 mit x bezeichnen; wir haben diesen Namen auch bereits in G3 geschrieben. Wir aktivieren B3:G3 und verfolgen dann das Menü (Excel 2010) Formeln/definierte Namen/Aus Auswahl erstellen. Zellen aktivieren heißt, dass die Zellen bei gedrückter linker Maustaste angefahren werden. Es erscheint das Fenster in Abb. 2.32 (T). Der Assistent hat bereits erkannt, dass in unmittelbarer Umgebung, nämlich in der rechten Spalte des aktivierten Bereichs, ein Name steht. Der Name entspricht unserer Absicht, und wir klicken OK. In anderen Excel-Versionen kann die Menüfolge anders sein. Fragen Sie in der Excel-Hilfe unter Namen aus Auswahl erstellen nach!

Wir benennen den Matrixbereich B4:F8, indem wir ihn aktivieren und dann Formeln/Definierte Namen/Namen definieren aufrufen. Es erscheint das Fenster Neuer Name. Der Bereich Bezieht sich auf: ist schon ausgefüllt, weil wir ihn vor



Abb. 2.32 (T) Die Werte im Bereich B4:F8 werden aus dem Zeilenbereich B3:F3 und dem Spaltenbereich A4:A8 sowie aus dem Wert der Zelle $A$1 im Tabellenblatt „E.x“ berechnet, mit gemischten Zellbezügen in B5 und benannten Bereichen in D6. Im Dialogkasten wird B3:F3 mit dem Namen in G3 versehen (☑ Rechter Spalte)

dem Aufruf aktiviert haben. Wir müssen dann nur noch das Feld Name ausfüllen, in unserem Fall mit „r.0“.

Im Fenster Formeln/Definierte Namen/Namens-Manager erhalten wir einen Überblick über alle benannten Bereiche, siehe Abb. 2.33.

Die Namen gelten in der gesamten Arbeitsmappe, wenn sie zum ersten Mal vergeben werden. Die Namen „x“ und „y“ tauchen zweimal auf. Als sie zum ersten Mal im Tabellenblatt „Dist“ vergeben wurden, galten sie in der gesamten Arbeitsmappe, wie in der Spalte Bereich zu sehen ist. Als sie zum zweiten Mal im Tabellenblatt „MMult“ vergeben wurden, wurde ihr Geltungsbereich auf dieses Tabellenblatt beschränkt. Dort gelten dann natürlich die anderen Definitionen von x und y nicht.

▶ **Tim** Welche Definitionen gelten dann im Tabellenblatt „E.x“?

▶ **Mag** Das können Sie durch Probieren herausbekommen.

Zellenweiser Aufruf der Namen

In Abb. 2.34a (T) und b (T) wird das elektrische Feld einer Punktladung bei $(x_1, 0)$ in der xy-Ebene berechnet. Sowohl die Definition der Variablen und Konstanten als auch die Berechnung sind auf zwei Tabellenblätter, „Dist“ und „E.x“, verteilt.

Im Tabellenblatt „Dist“ wird der Abstand r_0 des Aufpunkts (x,y) zur Ladung bei $(x_1,0)$ ermittelt:

$$r_0(x,\, y) = \sqrt{(x - x_1)^2 + y^2}$$

Im Tabellenblatt „E.x“ wird die x-Komponente E_x des elektrischen Feldes berechnet:

$$E_x(x, y) = \frac{x - x_1}{r_0^3}$$



Abb. 2.33 Der Namens-Manager listet alle Namen und die zugehörigen Bereiche auf

Die in beiden Tabellenblättern benannten Bereiche (Vektoren x und y, Matrix r_0, Konstante x_1) können in jedem Tabellenblatt *zellenweise* mit ihren Namen aufgerufen werden, allerdings nicht im gesamten Bereich eines Tabellenblattes, sondern nur in dem Bereich, der mit dem Bereich im Tabellenblatt deckungsgleich ist, für den der Name vergeben wurde. Außerhalb dieses Bereichs kommt es zu Fehlermeldungen, siehe Abb. 2.34b (T). Der Zeilenvektor x kann bei den aktuellen Festlegungen seiner Koordinaten nur in den Spalten B bis F eines Tabellenblattes aufgerufen werden, der Spaltenvektor y nur in den Zeilen 4 bis 8 und die Matrix r_0 nur im Bereich B4:F8. Die Konstante x_1 bezeichnet nur eine einzige Zelle und kann in der gesamten Datei ohne Einschränkung aufgerufen werden.

Aufruf der Namen in einer Matrixfunktion

Man kann in einem beliebigen Bereich mit der richtigen Größe an beliebiger Stelle eines Tabellenblattes auf den Matrixbereich „r.0“ zugreifen, wenn man die Formeln als Matrixformeln eingibt, siehe Abb. 2.35 (T). Dazu aktiviert man den Bereich, der beschrieben werden soll, schreibt in die Funktionszeile die gewünschte Formel, in unserem Fall [=(x − x.1)/r.0^3], und schließt mit dem Zaubergriff ab. Das wurde in Abb. 2.35 (T) im Bereich I16:O21 gemacht.

	A	B	C	D	E	F	G	H
2		=WURZEL((B$3-E.x!$A$1)^2+$A5^2) =WURZEL((x-x.1)^2+y^2)						
3		-2	-1	0	1	2	x	
4	-2	2,9	2,3	2,0	2,2	2,8		
5	-1	**2,3**	1,5	1,0	1,3	2,1		
6	0	2,1	1,1	**0,1**	0,9	1,9		
7	1	2,3	1,5	1,0	1,3	2,1		
8	2	2,9	2,3	2,0	2,2	2,8		
9	y						r.0	

	A	B	C	D	E	F	G
1	**x.1**	0,1			**Tabellenblatt E.x**		
2				=(x-x.1)/r.0^3			
3		###	###	###	###	###	#WERT!
4		-0,1	-0,1	0,0	0,1	0,1	#WERT!
5		-0,2	-0,3	-0,1	0,4	0,2	#WERT!
6		-0,2	-0,8	-100	1,2	0,3	#WERT!
7		-0,2	-0,3	-0,1	0,4	0,2	#WERT!
8		-0,1	-0,1	0,0	0,1	0,1	#WERT!
9							

Abb. 2.34 **a** (links, T) Tabellenblatt „dist“ wie in Abb. 2.32 (T); der Abstand zu einem Punkt (x_1, 0) wird berechnet; x_1 wird im Tabellenblatt „E.x“ definiert. **b** (rechts, T) Tabellenblatt „E.x“. Die x-Komponente des elektrischen Feldes einer Punktladung bei (x_1; 0) wird berechnet. In Reihe 3 und in Spalte G ist der Index außerhalb des erlaubten Bereichs (B4:F8, siehe a)

	H	I	J	K	L	M	N	O	P
15				=(x-x.1)/r.0^3 als Matrixformel					
16		-0,1	-0,1	0,0	0,1	0,1	#NV	#NV	
17		-0,2	-0,3	-0,1	0,4	0,2	#NV	#NV	
18		-0,2	-0,8	###	1,2	0,3	#NV	#NV	
19		-0,2	-0,3	-0,1	0,4	0,2	#NV	#NV	
20		-0,1	-0,1	0,0	0,1	0,1	#NV	#NV	
21		#NV	#NV	#NV	#NV	#NV	#NV	#NV	
22									

Abb. 2.35 (T) Aufruf von benannten Variablen in einer Matrix

▶ Zaubergriff zum Abschluss von Matrixfunktionen: Ψ *STRG + Umschalt + Eingabe*

Im Bereich I16:O21 wird auf den Matrixbereich r_0 im Tabellenblatt „Dist" (Abb. 2.34a (T)) zugegriffen. Sinnvolle Werte entstehen nur im Bereich I16:M20, der der Größe von r_0 entspricht.

Mathematische Matrixoperationen, wie z. B. die Multiplikation eines Vektors mit einer Matrix, werden in Abschn. 5.8 behandelt.

2.9 Fragen zu Kurvenscharen

Zellbezüge

1. Was besagt die Besenregel Ψ *Der Dollar macht's absolut*?
2. Welche Formel muss in Zelle B5 der Abb. 2.36 (T) geschrieben werden, mit absoluten und relativen Bezügen, damit durch Kopieren dieser Formel in den Bereich B5:E25 vier Sinusfunktionen entstehen?
3. Welche drei Bereiche müssen wie benannt werden, damit die Formel in G5 im gesamten Bereich B5:E25 definiert ist, wenn sie dort eingeschrieben wird?

	A	B	C	D	E	F	G	H
1		1	2	3	4	Amplitude		
2		4	3	2	1	Kreisfrequenz		
3								
4		F.1	F.2	F.3	F.4			
5	0						*=A*COS(w*t)*	
6	1							
25	20							

Abb. 2.36 (T) Γ-Aufbau einer Tabelle zur Darstellung von vier Sinusfunktionen



Abb. 2.37 **a** (links, T) Eine Gerade wird in der Tabelle definiert und im Diagramm dargestellt. **b** (Mitte) Standarddiagramm der Daten in a. **c** (rechts) Dialogfeld ENTWURF/DATEN AUSWÄHLEN/ DATENREIHE/DATENREIHE BEARBEITEN, mit dem die Datenreihe x, y aus a in das Diagramm eingefügt wird

Diagramme

In Abb. 2.37a (T) sehen Sie Datenreihen x und y, in Abb. 2.37b das zugehörige Diagramm und in Abb. 2.37c das Dialogfeld, mit dem die Datenreihe in dieses Diagramm eingefügt wurde.

4. In welchen Tabellenbereichen stehen REIHENNAME, WERTE DER REIHE <u>X</u> und WERTE DER REIHE <u>Y</u>?
5. Wie wurde der Ausdruck $y = 1x + 2$ in der Tabelle erzeugt und wie wurde er als Legende in das Diagramm eingefügt?

Schieberegler

In Abb. 2.38 (T) sehen Sie zwei Einstellungen eines Schiebereglers.

6. Welches ist die verknüpfte Zelle (LINKED CELL)?
7. Welches sind die minimalen und maximalen Werte des Schiebereglers?
8. Die Formel in F27 greift auf die Zelle D27 zu. Wie lautet sie?
9. In einer Zelle steht die Formel =(A5 − 500)/100, um mit einem Schieberegler Dezimalzahlen zwischen −5 und 5 mit dem Abstand 0,1 zu erzeugen. Welches ist die verbundene Zelle (LINKED CELL) und welche Werte stehen für MIN und MAX des Schiebereglers?

Polarkoordinaten

10. Die Koordinaten eines Kreises werden am besten in Polarkoordinaten mit dem Winkel ϕ und dem Radius r angegeben. Wie erhält man die kartesischen Koordinaten x und y, die für ein xy-Diagramm gebraucht werden?



Abb. 2.38 (T) Zwei Einstellungen eines Schiebereglers



	B	C	D	E
11	**dphi**		**r.K**	
12				
13	**phi**	**x**	**y**	
14	0,00	10,00	0,00	
15	**0,52**	8,66	5,00	
16	1,05	5,00	8,66	
17	1,57	0,00	10,00	
24	5,24	5,00	-8,66	
25	5,76	8,66	-5,00	
26	**6,28**	10,00	0,00	

Abb. 2.39 **a** (links) Darstellung eines Kreises mit 12 Strecken. **b** (rechts, T) Koordinaten für den Kreis in a

Die Figur in Abb. 2.39a wurde durch die Tabellenorganisation in der Abb. 2.39b (T) erzeugt. Die Zellen C11 und E11 wurden mit den links danebenstehenden Namen versehen. Der Spaltenbereich B14:B26 erhält den Namen *phi.*

11. Wie groß sind diese Zahlen?

Setzen Sie in den folgenden drei Antworten Namen ein, wenn sie definiert sind!

12. Welche Formeln stehen in den Zellen B15 und B26?
13. Welche Formel steht in Spalte C unterhalb x?
14. Welche Formel steht in Spalte D unterhalb y?

Dopplereffekt und machscher Kegel

15. Ein Flugzeug befindet sich bei $t = 0$ am Ort (0,0). Es ist mit der Geschwindigkeit 800 km/h unter einem Winkel von 30° zur Horizontalen geflogen. Wo befand sich das Flugzeug zur Zeit $t = -5$ s (Ortsangaben für x und y in m)?
16. Eine Schallquelle befindet sich gegenwärtig (bei $t = 0$) an der Position (0; 0) und hat sich vorher mit konstanter Geschwindigkeit $v_S = 600$ km/h längs der x-Achse bewegt. Vor 10 s hat sie einen Schallpuls ausgesendet. Welchen Radius r_c hat der Wellenkamm dieses Schallpulses gegenwärtig und wo ist sein Mittelpunkt x_c?

Exponentialfunktion

17. Um eine Exponentialfunktion freihändig auf Papier zu zeichnen, beginnt man zweckmäßigerweise mit einer Geraden als Hilfslinie. Wie wird diese Gerade durch die Parameter der Exponentialfunktion, Amplitude A und Zeitkonstante τ, bestimmt?
18. Eine Exponentialfunktion wächst von 1 auf 2, wenn das Argument von $t = 0{,}0$ auf 0,1 s erhöht wird. Wie stark wächst sie, wenn das Argument von 0,0 auf 0,2 s erhöht wird? Denken Sie binär! Wie stark wächst sie, wenn das Argument von $t = 0{,}8$ auf 1,0 s erhöht wird? Erstellen Sie zwei Skizzen von derselben Exponentialfunktion, jedes Mal mit der t-Achse von 0 bis 1,2, aber mit einer y-Achsen-Skalierung für die erste Skizze von 0 bis 4 und für die zweite



	A	B	C	D	E	F	G
1	**tau**	10		**8,44**	*=D4/C2*		
2	**A**	6	**5,70**	*=SUMME(Akt)*			
3				*=SUMMENPRODUKT(C7:C36;D7:D36)*			
4				**48,10**			Gerade
5	**t**	**f**	**Akt**	**t.m**		**t**	**y**
6	0	6,00					
7	1	5,43	0,57	0,5			
8	2	4,91	0,52	1,5			
35	29	0,33	0,03	28,5			
36	30	0,30	0,03	29,5			

Abb. 2.40 **a** (links) Restmenge einer radioaktiven Substanz. **b** (rechts, T) Daten für a

Skizze von 0 bis 16 × 4! Was sieht man in der zweiten Skizze, das in der ersten Skizze nicht ins Auge fällt?

Radioaktiver Zerfall

Die Kurven im Diagramm Abb. 2.40a wurden mit der Tabelle in Abb. 2.40b (T) erstellt.

19. Die Restmenge einer radioaktiven Substanz ist 130 bei $t = 10$ s und 110 bei $t = 12$ s. Wie groß ist die Aktivität in diesem Zeitabschnitt? Über welchem Zeitpunkt t_m sollte man diese Aktivität in einem Diagramm darstellen?
20. Wie lautet die Formel im Spaltenbereich B6:B36?
21. Welche Formeln stehen im Bereich F6:G7?
22. Wie groß ist die Fläche unter der Geraden in Abb. 2.40a (exakt, numerisch und formelmäßig)?
    Wie groß ist die Fläche unter der Exponentialfunktion in Abb. 2.40b (T) (geraten, formelmäßig)?
23. In D1 wird die mittlere Lebensdauer aus den numerischen Daten im Bereich C:D berechnet. Interpretieren Sie die Formeln im Bereich D:E!
24. In welcher Zelle wird eine gewichtete Summe und in welcher Zelle ein gewichteter Mittelwert berechnet?
25. Theoretisch wird erwartet, dass nach hinreichend langer Zeit die Werte in B1 und D1, sowie in B2 und C2 gleich groß sind. Begründen Sie diese Erwartung! Warum sind die experimentellen Werte D1 und C2 kleiner als erwartet?

Kosinusfunktionen

In Abb. 2.41a und b werden zwei Kosinusfunktionen dargestellt.

26. Wie groß sind die Amplituden und Periodendauern der oben dargestellten Funktionen Cos_1 und Cos_2?
27. Wie groß sind die Kreisfrequenzen der beiden Funktionen?
28. Was sind Obertöne zum Grundton mit der Frequenz $f = 100$ Hz?



Abb. 2.41 **a** (links) Kosinusfunktion Cos_1. **b** (rechts) Kosinusfunktion Cos_2

29. Wie Interpretieren Sie die Besenregel: Ψ *Cos plus Cos = Mittelwert mal halbe Differenz?*
30. Eine Kosinusfunktion wird zu einer zweiten Kosinusfunktion mit $f = 100$ Hz addiert. Welche Frequenz muss die erste Kosinusfunktion haben, damit eine Schwebung von 1 Hz entsteht?

Tabellenfunktion INDIREKT

31. Betrachten Sie die Einträge in acht Zellen einer Tabelle: A1 = 5; B2 = K; E1 = 5; K5 = 7; W1 = „E"&1; W2 = INDIREKT(W1); W3 = B2&A1; W4 = INDIREKT(W3). Welche Zahlen erscheinen in den Zellen W2 und W4?

http://www.springer.com/978-3-642-37856-0

Physik mit Excel und Visual Basic
Grundlagen, Beispiele und Aufgaben
Mergel, D.
2017, V, 372 S. 310 Abb., Softcover
ISBN: 978-3-642-37856-0